
Before Starting

This book is meant to be an active companion during the process of learning how to use
Mathematica. The main body of the text will certainly provide insights into how Mathemat-
ica works, but the examples should be retyped as a starting point for individual exploration.
Each chapter contains discussion, tips, and a description of Mathematica functionality, along
with actual examples that serve as starting points. Each chapter ends with additional exer-
cises to emphasize comprehension, which can be used as an assignment to students or simply
to work through on your own.

No matter what format this book is viewed in, it is recommended that readers have Mathe-
matica on the desktop or Mathematica Online immediately accessible to type the examples
and work through the exercises. It is recommended that as readers work through the book,
they save a new file for each chapter in Wolfram Notebook format (.nb), either locally or
in the Wolfram Cloud, for future reference.

All new Mathematica students should work through chapters one through twelve (at least)
to obtain the necessary basis of how to use Mathematica for solving mathematical problems
in different mathematical subjects. These chapters will be of value to intermediate Mathe-
matica users by filling in gaps in knowledge that can result from using Mathematica only for
a narrowly defined set of tasks, or by broadening the horizons of users who may have learned
Mathematica from an older version.

There is a lot more to Wolfram Language, like dealing with the vast majority of options of
commands and using new commands for topics and areas that we have missed, than we have
been able to cover in this book.

If you have understood what is in this book, and can do its exercises, then you can now
consider yourself a Wolfram Language programmer! There will always be more you can learn,
but you are ready to start using what you know to do real programming.

As a mathematician, there will probably be something you want to solve or program every
day. With a traditional computer language it would take too long to actually do it. But
with the Wolfram Language, and with all its built-in knowledge and automation, anyone
who knows the language can write very useful programs even in a matter of minutes. The
first step in creating a program for something is to see how to think about the thing in
computational terms. It might be something where computers have long been used. It
might be something that is only now conceivable for computers as a result of the Wolfram
Language. Whatever it is, try to imagine a Wolfram Language function for doing it. What
input would the function get? What output would it generate? What might the function be
called? Do not at first think about how you would write the code. Just think about what
the function should do. And only after you have understood that, start writing the code.

4

1 Introduction

1.1 What Is the Wolfram Language?

The Wolfram Language is a computer language. It gives you a way to communicate with
computers, in particular so you can tell them what to do. There are many computer lan-
guages, such as C++, Java, Python and JavaScript. The Wolfram Language is unique in
that it’s knowledge based. That means that it already knows a lot, so you have to tell it
much less to get it to do things you want.

It’s designed to make it as easy as possible to describe what you want, making use of
huge amounts of knowledge that are built into the language. And the crucial thing is that
when you use the Wolfram Language to ask for something, the computer immediately knows
what you mean, and then can actually do what you want.

You can make things that are visual, textual, interactive or whatever. You can do analyses
or figure things out. You can create apps and programs and websites. You can take a very
wide variety of ideas and implement them-on your computer, on the web, on a phone, on
tiny embedded devices and more.

1.2 Wolfram Cloud

The Wolfram Open Cloud gives anyone a way to do “casual” programming whenever they
want-with access to all the core computation, interface, deployment and knowledge capa-
bilities of the Wolfram Language. If you want to get more serious-about computation,
deployments or storage-you’ll need to have an actual subscription for Wolfram Cloud. With
the introductory Wolfram Cloud plan, you get 200 MB of cloud storage and temporary
file storage and deployments (each file and deployment expires after 60 days). For more
information, visit support.wolfram.com.

To start using your Wolfram Cloud product: in a
browser, go to wolframcloud.com. Its homepage
appears.

To create a new account, click sign up for free.
Enter your email address, first name, last name,
and password, and agree to the terms of service
and privacy policy.

To access your Wolfram Cloud product, click Sign
in if you have an account. Its login screen appears.
Enter your Wolfram ID and password, then select
Sign in. The product’s starting page appears. En-
joy with Mathematica

1.3 What is Mathematica?

Mathematica is a tool for technical computing based on the Wolfram Language. It contains
an extensive knowledge base for working with a very broad range of tasks, including solving
equations, programming, importing and exporting data, visualizing functions and data, and
much more.

Although Mathematica is a very large and powerful system, you can get up and running
with it in just a few minutes and become fairly proficient by learning some basic concepts
about how it is organized, the syntax of its commands, and how to get help when you
are stuck. This course will guide you through these first steps and provide you with the
foundation you need to incorporate Mathematica into your work and/or teaching.

Calculations can often be written in several different styles, with advantages and disad-
vantages in each scenario. This book focuses on conventions and shortcuts in the Wolfram
Language to make calculations shorter, clearer, or easier to understand.

1.4 The Structure of Mathematica

The basic parts of the Mathematica system:

1. Mathematica kernel the part that actually performs computations.

2. Mathematica front end the part that handles interaction with the user.

The most common way to work on Mathematica is to use interactive documents known
as notebooks. Notebooks mix Mathematica input and output with text, graphics, palettes,
and other material. You can use notebooks either for doing ongoing computations, or as
a means of presenting or publishing your results. You should realize that notebooks are
part of the “front end” to Mathematica. The Mathematica kernel which actually performs
computations may be run either on the same computer as the front end, or on another
computer connected via some kind of network or line. In most cases, the kernel is not even
started until you actually do a calculation with Mathematica.

1.5 Notebook Interfaces

If you use your computer via a purely graphical interface, you will typically double-click
the Mathematica icon to start Mathematica. In a “notebook” interface, you interact with
Mathematica by creating interactive documents. The notebook front end includes many
menus and graphical tools for creating and reading notebook documents and for sending
and receiving material from the Mathematica kernel. A notebook mixing text, graphics, and
Mathematica input and output.

6

When Mathematica is first started, it displays an empty notebook with a blinking cursor.
You can start typing right away. Mathematica by default will interpret your text as input.
You enter Mathematica input into the notebook, then press Shift + Enter together to make
Mathematica process your input. If your keyboard has a numeric keypad, you can use its
Enter key instead of Shift + Enter . After you send Mathematica input from your notebook,
Mathematica will label your input with In[n]:=. It labels the corresponding output Out[n]=.
Labels are added automatically.

The output is placed below the input. By default, input/output pairs are grouped using
rectangular cell brackets displayed in the right margin. In addition to the standard textual
input, Mathematica supports the use of generalized, non-textual input such as graphics and
user interface controls, freely mixed with textual input. To exit Mathematica, you typically
choose the Exit menu item in the notebook interface.

Important Notes When you input a command in Mathematica, make sure you do the
following:

1. Use upper and lower case characters exactly as we do. Mathematica is very “case
sensitive”. If you use the wrong capitalization, you may not get the desired result.

2. Use exactly the type of brackets we show. There are three types of brackets: square
brackets [ ], parentheses ( ), and curly braces { }. Each has its own meaning in
Mathematica. If you use the wrong one, Mathematica may not do what you expect.

7

3. Your output might appear in a slightly different from ours in some examples. We
explain why in the discussion of each example.

4. If your Mathematica notebook contains a lot of output, especially if graphical output is
involved, it can become very large when saved to your disk. Consider choosing Delete
All Output from the Cell menu before saving your notebook to disk. Your saved file
will then be much smaller, and it will be easier to transmit electronically to others
using Mathematica.

5. You can stop Mathematica in the middle of a computation by choosing Abort Eval-
uation under the Evaluation menu.

1.6 Palettes

Palettes are specially-prepared types of notebooks that provide graphical shortcuts for enter-
ing commands and expressions for those who like visual menus. You open one of them by
using the Palettes Menu. There are several predefined palettes available, depending on the
specific version of Mathematica you have. For example, The Basic Math Assistant palette
lets you enter expressions involving integrals, roots, and fractions in a more pleasing, Math-
ematical way. It also provides buttons to enter Greek symbols and some special characters
directly.

Many of the buttons on the Basic Math Assistant palette, as well as the other Assistant
palettes available from the Palettes menu, provide command templates when they are clicked.
For example, navigating to the 2D tab of the Basic Commands section and clicking the Plot
button yields the following.

Plot[ function , { var , min , max } ]

Such a template provides the appropriate syntax for the command name and only requires
the user to enter the remaining arguments before evaluating the command. The arguments

8

can be entered with the keyboard (and Tab can be used to jump between the placeholders)
or by clicking buttons in the palette.

You can use the CreatePalette command to construct your own custom palette, which
is handy if you find yourself doing the same operations and typesetting constructions over
and over again.

1.7 Important Resources

Stephen Wolfram Book Online Complete text, with full runnable examples and automat-
ically graded exercises.
wolfram.com/language/elementary-introduction/2nd-ed/

Wolfram Language Home Page Broad collection of resources about the Wolfram Lan-
guage.
wolfram.com/language/

Wolfram Documentation Center Documentation on all functions in the Wolfram Lan-
guage, with extensive examples.
reference.wolfram.com/language/

Wolfram Programming Lab Online and desktop access to the Wolfram Language, with
educational Explorations.
wolfram.com/programming-lab/

Wolfram Challenges Dynamic collection of online Wolfram Language programming chal-
lenges.
challenges.wolfram.com

2 Mathematica as a Calculator

2.1 Commands for Basic Arithmetic

Mathematica works much like a calculator for basic arithmetic. Just use the +, -, *, and /

keys on the keyboard for addition, subtraction, multiplication, and division. As an alternative
to typing *, you can multiply two numbers by leaving a space between them (the × symbol
will automatically be inserted when you leave a space between two numbers). You can raise
a number to a power using the ^ key. Use the dot . to type a decimal point.

Example 2.1. Calculate:

1)
25.5

5

In[ ]: 25.5 / 5

Out[ ]: 5.1

2) 4+ 25

In[ ]: 4 + 2^5

Out[ ]: 36

3)
23

5
− 3

5
+ 5

(
23
)

In[ ]: 23/5 - 3/5 + 5 * 2^3

Out[ ]: 44

Did you notice that when you entered the expressions in Example (2.1), Mathematica was
actively coloring parts of your input as you typed? Mathematica uses this coloring aid scheme
to tell the user whether the input is complete and syntactically correct.

2.2 Precedence

Mathematica follows the laws of precedence of multiplication over addition and so on, just
as you do by hand. Precedence of common operators is generally defined so that higher-level
operations are performed first. For simple expressions, operations are typically ordered from
highest to lowest in the order:

1. Parentheses.

2. Factorial.

3. Exponentiation, from right to left.

4. Multiplication and Division, from left to right.

5. Addition and Subtraction, from left to right.

10

Note Use parentheses ( ) to group terms in expressions. Do not use square brackets [ ]
or curly braces { }, they mean something different.

Example 2.2. In Mathematica, the expression
3× 5

4− 2
− 2

1+ 5
is entered as follows.

In[ ]: (3 * 5) / (4 - 2) - 2 / (1 + 5)

Out[ ]:
43

6

Note that, Mathematica normally gives you an exact (symbolic) value for every expression.

2.3 Built-in Constants

The Mathematical constants used most often are already built into Mathematica. The table
below shows a few of these. Notice that all names begin with a capital letter.

π Ratio of a circle’s circumference to its diameter Pi

e Natural exponential E

i Imaginary number I

◦ Degree to radian conversion multiplier Degree

∞ Positive infinity Infinity

Once again, you will see some coloring aids as you type the names of built-in constants.
When entering Degree, for example, the D will be colored black (it means something on
its own), then the characters will be colored blue until you type the final e, at which point
the entire name will be colored black. That’s because none of the names De, Deg, Degr, or
Degre is known. Also, use parentheses in expressions to clarify what you mean. This helps
avoid mistakes. For example, you might think that E^2Pi means e2π, but it doesn’t! It is
actually e2 × π because the exponentiation is done before the multiplication of 2 and π. To
get e2π, you should write E^(2Pi).

Note It is possible to enter each of these constants, or many other symbols, directly from
the keyboard, as well. You can type

Esc + p + i + Esc for π

Esc + e + e + Esc for e

Esc + i + n + t + Esc for
∫

Ctrl + / for ■
■

Ctrl + 6 for ■■

Ctrl + 2 for
√
■

Ctrl + 2 + 5 for ■
√
■

11

2.4 Built-in Functions

Mathematica has many built-in functions. These are the ones you will probably use the most.

Natural logarithm ln x Log[x]

Logarithm to base a loga x Log[a,x]

Exponential ex Exp[x]

Absolute value |x| Abs[x], RealAbs[x]
Square root

√
x Sqrt[x]

Trigonometric sin x, cos x, · · · Sin[x], Cos[x],..

Inverse trigonometric sin−1 x, cos−1 x, · · · ArcSin[x],..

Notes

1. The names of Mathematica’s built-in functions begin with an upper-case letter, and
each uses square brackets for the argument(s) of the function.

2. Mathematica uses radian measure for trigonometric functions.

Example 2.3. Evaluate:

1) sin

(
2π

3

)
In[ ]: Sin[2 Pi / 3]

Out[ ]:

√
3

2

2) log4 (1024)

In[ ]: Log[4, 1024]

Out[ ]: 5

3) cos (120◦)

In[ ]: Cos[120 Degree]

Out[ ]: −1

2

4)
√
2+

√
8

In[ ]: Sqrt[2] + Sqrt[8]

Out[ ]: 3
√
2

5) e3 ln 5

In[ ]: Exp[3 Log[5]]

Out[ ]: 125

6) tan−1(−∞)

In[ ]: ArcTan[-Infinity]

Out[ ]: −π

2

To evaluate the real-valued cube root of a real number x, use the CubeRoot[x] function.
We evaluate the value of 3

√
−2.46038 as follows.

In[ ]: CubeRoot[-2.46038]

Out[ ]: −1.35

12

In general, to evaluate the real-valued nth root of a real number x, use the built-in function
Surd[x,n].

Example 2.4. Evaluate 8

√
390625

6561
.

In[ ]: Surd[390625/6561, 8]

Out[ ]:
5

3

Notes

1. Log2[x] gives the base-2 logarithm of x. Also, Log10[x] gives the base-10 logarithm
of x.

2. You can add a comment to any expression by enclosing it within the symbol pairs (*
and *). For example:

In[ ]: 27*3 (* This multiplies 27 and 3. *)

Out[ ]: 81

Mathematica does not try to evaluate the phrase “This multiplies 27 and 3.” when
it evaluates your input. The phrase is for your use only. Mathematica reminds you of
this by coloring comments gray.

3. To do a computation, but do not print its output, you have to end your command by
semicolon ; .

2.5 Numerical and Scientific Notations

Mathematica normally gives you an exact (symbolic) value for every expression, for example

In[ ]: (3 + 9) * (4 - 8) / 1247 * 67

Out[ ]: −3216

1247

You can force Mathematica to give you an answer that looks like the decimal answer you
would get on a calculator by using the function N with square brackets around an expression. If
we attempt to give a result with n−digit precision for an expression (expr) we use N[expr,n].
For example,

In[ ]: N[(3 + 9)*(4 - 8)/1247*67]

Out[ ]: −2.57899

Example 2.5. Find the 45−digits precision of π.

In[ ]: N[Pi, 45]

Out[ ]: 3.14159265358979323846264338327950288419716940

13

Example 2.6. Find the numerical value of log2

(√
2+ 7

√
123

1+ sin
(
π
13

)) using 16−digits of precision.

In[ ]: N[Log2[(Sqrt[2] + Surd[123, 7])/(1 + Sin[Pi/13])], 16]

Out[ ]: 1.457204671921001

Mathematica uses standard scientific notation to display results when the numbers either get
very large or very small. For example,

In[ ]: N[1234567890]

Out[ ]: 1.23457× 109

In[ ]: N[1234567890,2]

Out[ ]: 1.23× 109

In[ ]: N[0.000003492836]

Out[ ]: 3.49284× 10−6

Note The command IntegerPart[x] gives the integer part of x while FractionalPart[x]
gives the fractional part of x. For example

In[ ]: IntegerPart[N[Pi, 10]]

Out[ ]: 3

In[ ]: FractionalPart[N[Pi, 10]]

Out[ ]: 0.141592654

2.6 Prefix and Postfix Forms for Built-in Functions

There are three ways to write expressions in Mathematica.

f[x,y] Standard form for f[x,y] N[Sqrt[2]] 1.41421
f@x Prefix form for f[x] N@Sqrt[2] 1.41421
x//f Postfix form for f[x] Sqrt[2]//N 1.41421

You should notice that // has very low precedence. If you put //f at the end of any expression
containing arithmetic or logical operators, the f is applied to the whole expression. So, for
example, x+y//f means f[x+y], not x+f[y]. While the prefix form @ has a much higher
precedence. f@x+y is equivalent to f[x]+y, not f[x+y]. You can write f[x+y] in prefix
form as f@(x+y).

2.7 Mathematica Help

In order to get a quick description of a command, use ?Command. For example, to get quick
info about the function N[] just type ?N.

If you need more explanation use ??Command. If you need even more, use Mathematica’s
Document Center in the Help Menu and search for the command. Mathematica comes with
an excellent Help which contains explanations and many nice examples demonstrating how
to use each of the functions available in this software.

Using the Input Assistant The Input Assistant helps you automatically complete code,
discover functions and options, and reduce oversights and typographical errors while coding.

As you type, a list of possible functions and variables (both system- and user-defined) is
displayed after a user-defined delay. The list is refined automatically as you type additional
characters.

Completions can be inserted using the keyboard or the mouse. Select a completion with the
mouse pointer or arrow keys. Press Enter or Tab , or click to insert the completion.

Access documentation for a function or variable by clicking the document icon � next to
the function name. The documentation will open in a new window.

Function Templates describe common formatting for specified functions. Function tem-
plates can be accessed via code completion. If an inserted function has associated templates,
a new button will display after inserting the completion.

15

Press Tab or Enter , or click the displayed down arrow icon to access a list of templates for
the current function. Alternatively, pressing Ctrl + Shift + K after fully typing a function
name will access the template list.

Inserted function templates are fully editable text. Variables that require completion are
represented by a yellow placeholder. The currently selected placeholder is highlighted in blue
and you can use the Tab key to advance to the next placeholder. Typing any character
will replace the placeholder with that character. You must replace each placeholder to
successfully evaluate your input.

Note The Input Assistant feature set has many other components such as:

1. Option templates: View and insert options related to your current function.

2. Dynamic highlighting: Highlight code as you type to more easily identify which part
of the code you are working in.

3. String completion: Autocomplete string arguments inside functions.

4. Color chooser: Choose a specific RGB color from an autocomplete interface.

Wolfram Predictive Interface Once you finish a computation, the Suggestions Bar pro-
vides immediate access to possible next steps optimized for your results. Clicking suggested
actions will perform the action on the output.

16

2.8 Exercises

1. Compute a 20 decimal place approximation to e, the base of the natural logarithm.
Ans. 2.7182818284590452354

2. Calculate

(a)

∣∣∣∣∣ log2(128) + 6
√
16777216

cos
(
sin−1

(
1
2

)) ∣∣∣∣∣ (b) tan
(π
5

)

Ans. (a)
46√
3

(b)
√
5− 2

√
5

3. Compute the common logarithm (base 10) of 25. What is its numerical approximation?
Ans. 1.50515

4. What happens if you try to subtract ∞ from ∞? What happens if you compute
1

0
?

Ans. Indeterminate, ComplexInfinity

5. Can Mathematica calculate the expression 29941 − 1 easily?
Ans. Yes

6. Show that

√
3
√
64
(
22 +

(
1
2

)2)− 1 = 4.

7. Add parentheses to 4− 2 ∗ 3+ 4 to make 14.

8. Try to find the numerical value of sin2
(π
5

)
.

Ans. 0.345492

9. Use the help to read about the command FunctionExpand, and use it to prove that

sin (24◦) =
1

8

√
3
(√

5 + 1
)
− 1

4

√
1

2

(
5−

√
5
)
.

10. Find the rational number with smallest denominator that approximates the value of π
with error (tolerance) 0.001.
Hint: see the command Rationalize.

Ans.
201

64

3 Variables and Functions

3.1 Rules for Names

You are free to make up the names of the variables you use, as long as you use only letters
and numbers and obey these rules:

1. Names cannot begin with a number.

2. Names cannot contain special characters like @, -, , &, %,!.

3. You cannot use names that conflict with the names that are predefined inMathematica.
For example, you cannot name one of your own variables Sin.

Notice that because of the first rule above, Mathematica will automatically “do the right
thing” with an expression such as 2x. Since this is not an acceptable name, Mathematica
will interpret it as “2” times “x”. All of the following are examples of legitimate names that
you can use for variables:

a, m, pI, A, area, perimeter, ABBA, good4you, classOf2016

Mathematica distinguishes uppercase and lowercase characters. For example, the names
Batman, batman, and batMan are all different. One convention we will use throughout the
remainder of the text is that all of the variable names we define will begin with a lowercase
letter. You will know that the names belong to us, since all of Mathematica’s predefined
names start with an uppercase letter.

3.2 Immediate Assignment

You can define a name by assigning any value to it. You can then use the value whenever
you want later in a computation. You do this using the equal sign =, which is the symbol
for immediate assignment. For example:

In[ ]: a = 3

Out[ ]: 3

The name “a” can also be called a symbol or a variable. Once you have assigned a value to
a variable, you can recall it by using it directly in an expression:

In[ ]: N@Sqrt[2/a]

Out[ ]: 0.816497

You may want to know how Mathematica keeps track of all the symbols and variables that
you define. Say we assign the name mySum to the sum of x and 3 times y.

In[ ]: mySum = x + 3 y

Out[ ]: x+ 3y

In this case, Mathematica simply repeated our definition because x and y do not yet have
an associated value. Now suppose we assign the value 2 to x and the value 5 to y, and then
reevaluate mySum:

In[ ]: x = 2; y = 5; mySum

Out[ ]: 17

When Mathematica reevaluated mySum this time, it replaced x and y by their respective
values, and simplified the resulting expression.

Note Mathematica will interpret spaces between letters as multiplication, but it will not
put a × sign on the screen. If you are not looking carefully, you may not notice the difference
between the expression x y and xy. The former means x times y, while the latter is the
name of a single variable called xy.

Clearing Symbols You tell Mathematica to forget about the assignment using the Clear
command:

In[ ]: Clear[a]

Out[ ]: a

You can also Clear assignments for many names at the same time in one statement
Clear[mySum, x, y, X, Y, A, B]. You can use the command

In[ ]: Clear["Global’ *"]

to clear all the variable and function names you have created so far in your Mathematica
session.

3.3 Functions

Mathematica has many built−in functions such as N, Sqrt, Sin and Log. You can add your
own functions as well.

Defining Functions To define a function f(x) in Mathematica, you use the syntax:

f[x ] := formula in terms of the variable x

This syntax may look a little awkward, but you should notice:

� The underscore character immediately following the variable on the left, x_, tells Math-
ematica that x is the variable of the function.

� The colon−equal sign := is a delayed assignment command. It behaves differently
from =. The basic difference between these forms is when the expression at rhs is
evaluated. lhs=rhs is an immediate assignment, in which rhs is evaluated at the time
when the assignment is made. lhs:=rhs, on the other hand, is a delayed assignment,
in which rhs is not evaluated when the assignment is made, but is instead evaluated
each time the value of lhs is requested.

� After you type the x_, Mathematica visually identifies occurrences of the variable name
in the formula by setting them in italic type. On some systems, the variable will also
appear in a different color, like the green color.

Example 3.1. Define the function f(x) =
x2 + 4

x− 1
in Mathematica, and then evaluate the

value of f(3), f(−1.2), and f(1).

In[ ]: f[x ] := (x^2 + 4)/(x - 1)

In[ ]: f[3]

Out[ ]:
13

2

In[ ]: f[-1.2]

Out[ ]: −2.47273
In[ ]: f[1]

Out[ ]: ComplexInfinity

Functions of Several Variables Functions with more than one variable are defined using
a similar syntax, as illustrated in the following examples.

Example 3.2. Define a function that computes the average speed of a moving object passes
a distance 40 (km) in a time 34 (min).

In[ ]: speed[distance , time ] := distance / time

In[ ]: N@speed[40, 34]

Out[ ]: 1.17647 (* 1.17647 kilometers per minute. *)

In[ ]: N@speed[40, 34/60]

Out[ ]: 70.5882 (* 70.5882 kilometers per hour. *)

Piecewise Functions A new command named Piecewise has been introduced in Math-
ematica 6 specifically for dealing with functions of split definition. Its syntax has either of
these two forms, each of which pairs together values and conditions:

Piecewise[{ {val1, cond1} , {val2, cond2} , · · ·}]

Piecewise[{ {val1, cond1} , {val2, cond2} , · · ·}, default value]

Example 3.3. Let g(x) =


1− x : 0 < x ≤ 2

x ln x : 2 < x ≤ 5

ex : x ≤ 0 or x > 5

. Find the values of g(1), g(5), and

g(−4).

In[ ]: g[x ]:=Piecewise[{{1-x, 0<x<=2}, {x Log[x], 2<x<=5}}, Exp[x]]

In[ ]: g[1]

Out[ ]: 0

In[ ]: g[5]

Out[ ]: 5 ln 5

In[ ]: g[-4]

Out[ ]: e−4

Use Esc + p + w + Esc to enter { and Ctrl + , and then Ctrl + Enter for each additional
piecewise case.

3.4 Transformation (Substitution) Rule

You can substitute values into any symbolic expression without having to assign values to
the variables explicitly. The substitution symbol “/.” is made using the slash and period
symbols, with no space in between. It is used in the form:

expression /.
{
list of transformations using ->, separated by commas

}
For example, to substitute x = 2 and y = 5 into the expression x2 − 2xy:

In[ ]: x^2 - 2*x*y /.{x->2, y->5}
Out[ ]: −16

The arrow symbol -> is formed by entering the minus sign and greater than sign together,
with no spaces in between. (As soon as you finish typing these two signs, Mathematica will
automatically change them to a very spiffy looking →) The arrow symbol -> represents a
transformation rule. We may use it many times throughout the lectures. The primary ad-
vantage of using the transformation command is that the value you substitute into a variable
is temporary and is not assigned to the variable. It is not remembered by Mathematica.

Example 3.4. The roots of a quadratic function f(x) = ax2 + bx+ c are given by the
quadratic formula

r1 =
−b+

√
b2 − 4ac

2a
and r2 =

−b−
√
b2 − 4ac

2a

Let f(x) = x2 + x− 6. Find the larger root of f.

In[ ]: (-b + Sqrt[b^2 - 4 a c])/(2 a) /. {a -> 1, b -> 1, c -> -6}
Out[ ]: 2

Note You can use /. to replace heads of expressions:

In[ ]: Sin[x] /. Sin -> Cos

Out[ ]: cos(x)

3.5 Anonymous Functions

Sometimes we need to “define a function as we go” and use it on the spot. Mathematica
enables us to define a function without giving it a name (nor any reference to any specific
variables) use it, and then move on! These functions are called anonymous or pure functions.
Obviously if we need to use a specific function frequently, then the best way is to give
it a name and define it as we did before. Here is an anonymous function equivalent to
f(x) = x2 + 4.

In[ ]: (#^2+4)&

The expression (#^2+4)& defines a nameless function. As usual we can plug in data in
place of #. The symbol & determines where the definition of the function is completed. For
example, we find the value of f(x) = x2 + 4 when x = 5 using the pure function form as
follows

In[ ]: (#^2+4)&[5]

Out[ ]: 29

Also, the value of g(x) =
√
x sin(x) when x =

π2

16
in numerical form is

In[ ]: (Sqrt[#] Sin[#])&[Pi^2/16] //N

Out[ ]: 0.454328

Anonymous functions can handle several variables. Here is an example of an anonymous
function for f(x, y) =

√
x2 + y2.

In[ ]: Sqrt[#1^2+#2^2]&[3,4]

Out[ ]: 5

As you might guess, #1 and #2 refer to the first and second variables in the function.

3.6 Functions with Conditions

Everything in Mathematica is an expression and each expression has a pattern. One can
search for a specific pattern and change it to another pattern. This is called pattern matching
programming.

The Wolfram Language provides a general mechanism for specifying constraints on pat-
terns. All you need to do is to put /;condition at the end of a pattern to signify that it
applies only when the specified condition is True. For example, this gives a definition for f
that applies only when its argument x is positive:

In[ ]: f[x /; x > 0] := x^2

In[ ]: f[5]

Out[ ]: 25

In[ ]: f[-4]

Out[ ]: f[−4]

In general, you can put /;condition at the end of any := definition to tell the Wolfram
Language that the definition applies only when the specified condition holds.

In[ ]: f2[x ] := x^2 /; x > 0

In[ ]: f2[5]

Out[ ]: 25

In[ ]: f2[-4]

Out[ ]: f2[−4]
There is a collection of functions built into the Wolfram Language for testing the properties of
expressions. Some of these functions are: Positive, Negative, EvenQ, OddQ and PrimeQ.
Using the function Positive, and the question mark "?" as a pattern test, we can define:

In[ ]: f3[x ?Positive] := x^2

In[ ]: f3[5]

Out[ ]: 25

In[ ]: f3[-4]

Out[ ]: f3[−4]
Note that the condition expr /;test can be shortened to expr ?test

Types of Numbers Four underlying types of numbers are built into the Wolfram System:
Integer, Rational, Real and Complex.

Example 3.5. Define the Collatz function f(x) =

{x

2
: x is even

3x+ 1 : x is odd
.

In[ ]: f[x Integer?EvenQ] := x/2

f[x Integer?OddQ] := 3 x + 1

3.7 Exercises

1. The arc length of a segment of a parabola ABC of an ellipse with semi−minor axes a
and b is given approximately by:

LABC =
1

2

√
b2 + 16a2 +

b2

8a
ln

[
4a+

√
b2 + 16a2

b

]
Determine LABC if a = 11 cm and b = 9 cm.

Ans. 24.5637

2. Define the two functions f(x) = x2 − 1 and g(x) = x3.

(a) Evaluate: f
(
g(x)

)
, g
(
2f(3)− 13

)
, f
(
f(f(0.5)))

)
.

Ans. x6 − 1 , 27 , −0.808594
(b) What will the following functions do?

Compose[f, g, x]

ComposeList[{g, f}, x]

3. In the triangle shown a = 5.3 cm, γ = 32◦, and b = 6 cm. Define α, β, and c as
variables, and then:

(a) Calculate the length c by using the Law of Cosines:

c2 = a2 + b2 − 2ab cos γ.

Ans. 3.18656

(b) Calculate the angles α and β (in degrees) using the Law of Cosines.
Ans. α = 61.8095◦ , β = 86.1905◦

4. Define the function g(x) =

{
x sin(πx) : −π ≤ x ≤ π

1− x : otherwise
, then evaluate the value of

g(−5), g(6), g(π), and g(1.8).
Ans. 6,−5, π sin

(
π2
)
,−1.05801

4 Lists

4.1 What is a List?

A list in Mathematica is an expression whose elements are separated by commas and enclosed
in { curly braces }. The elements of a list can be of any type, and need not be all of the
same type. For example:

{2, 5, 7, 10, -3, -25}

{"good", 17, {23, 89}}

Lists are important structures in Mathematica. Many of Mathematica’s inputs and outputs
are expressed using lists.

Lists respect order, so that {1,2} is not the same as {2,1} in Mathematica. Also, a list
can contain a copy of the same object several times, so {1,2,1} is defer from {1,2}.

One of the nice features of Mathematica is that it often allows us to perform operations
simultaneously on each element of a list with a natural syntax, especially when the elements
of the list are numerical.

Example 4.1. For the list of the first five positive integers do the following.

1. add 6 to each element.

In[ ]: {1, 2, 3, 4, 5} + 6

Out[ ]: {7, 8, 9, 10, 11}

2. multiply each element by −3.

In[ ]: -3 * {1, 2, 3, 4, 5}
Out[ ]: {−3,−6,−9,−12,−15}

3. rise each element to the power 5.

In[ ]: {1, 2, 3, 4, 5}^5
Out[ ]: {1, 32, 243, 1024, 3125}

4. divide 120 by each element.

In[ ]: 120 / {1, 2, 3, 4, 5}
Out[ ]: {120, 60, 40, 30, 24}

If two lists have the same number of elements, we can add, subtract, multiply and divide
them together element by element. For example

In[ ]: {1, 2, 3} + {3, 1, 4}
Out[ ]: {4, 3, 7}
In[ ]: {3, 4}^{2, 3}
Out[ ]: {9, 64}
In[ ]: {1, 2, 3} * {3, 1, 4}
Out[ ]: {3, 2, 12}

4.2 Functions Producing Lists

Mathematica provides us with commands for which the output is a list. These commands
have a nature of repetition and replace loops in procedural programming.

Range

Range[imax]

generates the list {1,2,...,imax}

Range[imin,imax]

generates the list {imin,...,imax}

Range[imin,imax,di]

uses step di

Table

Table[expr,{i, imax}]
generates a list of the values of expr when i runs from 1 to imax

Table[expr,{i, imin, imax}]
starts with i = imin

Table[expr,{i, imin, imax, di}]
uses step di

Table[expr,{i, {i1, i2, ...}}]
uses the successive values i1, i2,...

Table[expr, n]

generates a list of n copies of expr.

Example 4.2. Produce the list of

1. the first ten integers

In[ ]: Range[10]

In[ ]: Table[i,{i, 10}]
Out[ ]: {1, 2, 3, 4, 5, 6, 7, 8, 9, 10}

2. the first 10 even integers

In[ ]: Range[2,20,2]

In[ ]: 2*Range[10]

In[ ]: Table[2i,{i, 10}]
Out[ ]: {2, 4, 6, 8, 10, 12, 14, 16, 18, 20}

3. the reciprocal of the first 10 odd integers

In[ ]: 1/Range[1,20,2]

In[ ]: Table[1/(2 i - 1), {i, 10}]

Out[ ]:

{
1,

1

3
,
1

5
,
1

7
,
1

9
,
1

11
,
1

13
,
1

15
,
1

17
,
1

19

}
4.3 Displaying Lists

The default output form of a list, like its input form, uses the curly brace notation.

{1,2,3}

Several formatting functions are available for displaying lists in different forms. For exam-
ple, TableForm is useful for displaying nested lists (multi-dimensional data) in a simple
rectangular array.

In[ ]: TableForm[{1,2,3}]
Out[ ]: 1

2

3

In[ ]: {{"A","B","C"},{1,2,3}} // TableForm

Out[ ]: A B C

1 2 3

Another useful function for displaying nested lists is Grid. It contains numerous options
specifically for formatting tabular data.

In[ ]: Grid[{{"A","B","C"},{1,2,3}}]
Out[ ]: A B C

1 2 3

4.4 Working with Elements of a List

Mathematica lets you work with the elements of a list directly. You can do this using either
the Part command, or - as a shortcut - the double square brackets [[ ]] notation. For
example, suppose we’re given the following definition of a list:

In[ ]: mylist = {"good",17,{23,89}};

Then

In[ ]: mylist[[2]]

Out[ ]: 17

In[ ]: Part[mylist,3]

Out[ ]: {23, 89}
In[ ]: mylist[[3,2]]

Out[ ]: 89

Example 4.3. Consider the list of the 26-English letters from "a" to "z". You can use the
command CharacterRange to generate such list.

In[ ]: letters = CharacterRange["a", "z"]

Out[ ]: {a, b, c, d, e, f, g, h, i, j, k, l, m, n, o, p, q, r, s, t, u, v, w, x, y, z}

1. Obtain the second element from the end of the list letters.

In[ ]: letters[[-2]]

Out[ ]: y

2. Pick out the 3rd, 10th, and the last characters in the list letters.

In[ ]: letters[[{3, 10, -1}]]
Out[ ]: {c, j, z}

3. Find all the characters from the 11th character to the 21th character in the list
letters.

In[ ]: letters[[11 ;; 21]]

Out[ ]: {k, l, m, n, o, p, q, r, s, t, u}

4.5 Useful Functions

We will list many of the commands we additionally use to manipulate lists and their elements.
All of the following examples will be demonstrated using the list newlist generated by 25
random integers between 0 and 10 using seed = 5:

In[ ]: SeedRandom[5]; newlist = RandomInteger[10, 25]

Out[ ]: {0, 0, 3, 3, 4, 5, 1, 9, 1, 0, 10, 5, 1, 8, 0, 5, 9, 5, 3, 7, 3, 5, 8, 6, 7}

� Length reports the number of elements in a list.

In[ ]: Length[newlist]

Out[ ]: 25

� Sort will arrange the elements of the list in the natural increasing order of the elements.

In[ ]: newlist//Sort

Out[ ]: {0, 0, 0, 0, 1, 1, 1, 3, 3, 3, 3, 4, 5, 5, 5, 5, 5, 6, 7, 7, 8, 8, 9, 9, 10}

� Reverse will reverses the order of the elements in a list.

In[ ]: Reverse@Sort@newlist

Out[ ]: {10, 9, 9, 8, 8, 7, 7, 6, 5, 5, 5, 5, 5, 4, 3, 3, 3, 3, 1, 1, 1, 0, 0, 0, 0}

� Count will show how many times a particular item appears in a list.

In[ ]: Count[newlist, 3]

Out[ ]: 4

� Total gives the total (sum) of the elements in list.

In[ ]: Total@newlist

Out[ ]: 108

4.6 Listable Functions

There are times when we would like to apply a function to all the elements of a list. Suppose f
is a function and {a,b,c} is a list. We want to be able to “push” the function f inside the list
and get {f[a],f[b],f[c]}. Many of Mathematica’s built−in functions have the property
that they simply “go inside” a list. A function with this property is said to be listable. For
example, All the arithmetic functions are listable, also Sqrt is a listable function.

In[ ]: Sqrt@{1, 4, 9}
Out[ ]: {1, 2, 3}

Should all operations and commands be listable ? The answer is clearly “no.” The simplest
example of a command that should not be listable is the Reverse command.

In[ ]: Reverse[{{1, 2}, {9, 10}}]
Out[ ]: {{9, 10}, {1, 2}}

Indeed, if Reverse were listable, the output above would instead have be treated as {{2, 1},

{10, 9}}. The Reverse command should not be listable because it acts on the structure
of the list as a whole. It is not designed to act individually on each element of a list. This
observation brings us to the Map command.

Map The Map command can be used to force listability of any command/function and
allow it to act on each element of a list. This command has the form:

Map[ function name , list ]

As an example,

In[ ]: Map[Reverse, {{1, 2}, {9, 10}}]
Out[ ]: {{2, 1}, {10, 9}}

The equivalent shorthand to apply a function to a list is /@ as follows

In[ ]: Reverse/@{{1, 2}, {9, 10}}
Out[ ]: {{2, 1}, {10, 9}}

Example 4.4. Consider the list

newlist = {0,0,3,3,4,5,1,9,1,0,10,5,1,8,0,5,9,5,3,7,3,5,8,6,7}.

Suppose we want to count how many times the numbers 6 and 3 appear in the list, we will
get unexpected answer using Count:

In[ ]: Count[newlist, {6, 3}]
Out[ ]: 0

This is because Count searches inside the list newlist for an expression like {6, 3} which
does not exist. The correct way to do the job is as follows:

In[ ]: Count[newlist, #] & /@ {6, 3}
Out[ ]: {4, 1}

4.7 Vectors

Vectors in Mathematica are represented by lists. There is no need to specify if a particular
vector is a row vector or column vector; this makes things easier for the user and keeps the
focus on the result instead of Mathematical bookkeeping.

Vectors can be constructed explicitly or programmatically. To define a vector manually,
just create a list. For example, the vectors v = (2, 3) and u = (−1, 1, 2) are entered as

In[ ]: v = {2, 3}
Out[ ]: {2, 3}
In[ ]: u = {-1, 1, 2}
Out[ ]: {−1, 1, 2}

To define a vector programmatically, use Table.

In[ ]: Table[i^2, {i, 1, 10}]
Out[ ]: {1, 4, 9, 16, 25, 36, 49, 64, 81, 100}

Mathematical operations like addition +, subtraction -, multiplication *, division /, and
exponentiation ^ can operate on vectors of the same length. The operations are applied
element-wise, so the operation is performed on the first two elements of the vector, then the
operation is performed on the second two elements of the vector, and so on.

Example 4.5. Let u = (2, 3, 7) and v = (7, 1, 5). Find 5u− 2v

In[ ]: u = {2, 3, 7};v = {7, 1, 5};
In[ ]: 5 u - 2 v

Out[ ]: {−4, 13, 25}

The Dot Product You compute the dot product of two vectors using the Dot command:

In[ ]: Dot[{1, 2, 3},{-4, 7, 0}]
Out[ ]: 10

Sometimes new users expect the multiplication of two vectors to give the dot product instead
of performing element-wise multiplication.

The Cross Product To compute a cross product of two vectors in R3 (also known as the
vector product), you use the Cross command.

In[ ]: Cross[{1, 2, 3},{-4, 7, 0}]
Out[ ]: {−21,−12, 15}

Length and Angle of Vectors The length, or the norm, of the vector u = (a1, a2, · · · , an)
is

∥u∥ =
√
a21 + a22 + · · ·+ a2n

In Mathematica, we use the command Norm to find the vector length. For example,

In[ ]: Norm[{1, 2, 1}]
Out[ ]:

√
6

Also the angle between two vectors u and v can be found using the command VectorAngle.
For example,

In[ ]: VectorAngle[{1, 1}, {1, 0}]

Out[ ]:
π

4

In[ ]: VectorAngle[{1, 1, 0}, {1, 0, 2}]

Out[ ]: cos−1

(
1√
10

)

31



Feras Awad @2025 4 LISTS

4.8 Matrices

Like vectors, matrices can be comprised of any sort of expression: symbols, numbers, strings,
and images, and even mixtures thereof. Matrices are constructed by creating nested lists.
Small matrices can be manually entered by typing, and the simplest method is to create
a nested list in one-dimensional format using list notation by entering each row as a list,
beginning with the first row. That is, a matrix is entered in the form:

{ list of row 1 , list of row 2 , ... }

For example, the matrix

[
3 −4 7

−1 0 5

]
is entered in Mathematica as follows:

In[ ]: {{3, -4, 7}, {-1, 0, 5}}

We can use MatrixForm to display the result in two-dimensional format.

In[ ]: {{3, -4, 7}, {-1, 0, 5}} // MatrixForm

Out[ ]:

(
3 −4 7
−1 0 5

)
Matrices can also be entered with palettes like the Basic Math Assistant. The Basic

Commands section of the palette has a tab for matrix commands, including a button that will
paste an empty 2× 2 matrix into a notebook. There are buttons to add rows and columns to
newly created matrices, providing users with an interactive way to construct a template for
a larger matrix. For example, clicking the matrix button to create a 2× 2 matrix template
and then clicking the Add Row and Add Column buttons once each will create a blank 3× 3

matrix template as seen in the following figure.

Programmatic creation of matrices can be accomplished using the Table function as shown
in the following examples.

Example 4.6. Define a 3× 4 matrix
(
aij
)
where aij = i+ j.

In[ ]: Table[i + j, {i, 3}, {j, 4}] // MatrixForm

Out[ ]:

 2 3 4 5
3 4 5 6
4 5 6 7



Example 4.7. Write a code that generates the matrix


1 2 3 4 5
6 7 8 9 10
11 12 13 14 15
16 17 18 19 20
21 22 23 24 25

.
In[ ]: Table[j + 5 (i - 1), {i, 1, 5}, {j, 1, 5}]
In[ ]: Partition[Range[25], 5]

Note that the command Partition[list, n] partitions a list into nonoverlapping sublists
of length n.

4.9 Special Types of Matrices

There are special commands in Mathematica for matrices with special structures like identity
matrices, constant matrices, and the triangular matrices (upper, lower, and diagonal).

1. The command ConstantArray[c, n] generates a list (vector) of n copies of the
element c. While ConstantArray[c, {m, n}] generates an m× n array of nested lists
containing copies of the element c. For example,

In[ ]: ConstantArray[3, 10]

Out[ ]: {3, 3, 3, 3, 3, 3, 3, 3, 3, 3}
In[ ]: ConstantArray[10, {2, 3}] // MatrixForm

Out[ ]:

(
10 10 10
10 10 10

)
2. The identity matrix, or a unit matrix, of size n is the n× n square matrix with ones on

the main diagonal and zeros elsewhere. IdentityMatrix[n] gives the n× n identity
matrix. For example,

In[ ]: IdentityMatrix[2] // MatrixForm

Out[ ]:

(
1 0
0 1

)
3. A diagonal matrix is a matrix (usually a square matrix) in which the off-diagonal

elements are all zero. The main diagonal entries themselves may or may not be zero.
The command DiagonalMatrix[list] gives a matrix with the elements of list on
the leading diagonal, and 0 elsewhere. For example, the diagonal matrix of the first 4
positive even integrs is

In[ ]: DiagonalMatrix[2 Range[4]] // MatrixForm

Out[ ]:


2 0 0 0
0 4 0 0
0 0 6 0
0 0 0 8


4. A triangular matrix is a special kind of square matrix. A square matrix is called lower

triangular if all the entries above the main diagonal are zero. Similarly, a square matrix
is called upper triangular if all the entries below the main diagonal are zero. A triangular
matrix is one that is either lower triangular or upper triangular. A matrix that is both
upper and lower triangular is called a diagonal matrix.

Example 4.8. Construct the 3× 3 upper triangular matrix whose elements follow the

rule aij =

{
2i− j : i ≤ j

0 : i > j
.

In[ ]: a[i , j ] := 2i-j /; i<=j

In[ ]: a[i , j ] := 0

In[ ]: Table[a[i,j], {i, 3}, {j, 3}] // MatrixForm

Out[ ]:

 1 0 −1
0 2 1
0 0 3


To generate the lower triangular matrix, just change the inequality sings in aij.

In[ ]: a[i , j ] := 2i-j /; i>=j

In[ ]: a[i , j ] := 0

In[ ]: Table[a[i,j], {i, 3}, {j, 3}] // MatrixForm

Out[ ]:

 1 0 0
3 2 0
5 4 3


4.10 Basic Matrix Operations

Mathematica can do addition, subtraction, scalar multiplication, and matrix multiplication
with the operators +, -, *, and ., respectively, in addition to many other matrix computations
like the determinant, the transpose, and the inverse. To see how Mathematica operates with
these computations, let

A =

[
1 2
3 6

]
B =

[
5 6
7 8

]
F =

 1 2
3 4
5 6

 G =

 1 2 3
4 5 6
7 8 9


and define them in Mathematica as a two-dimensional lists as follows. Care must be taken,
however, not to use //MatrixForm in the definition of the matrix. This command is for
display purposes only.

In[ ]: A = {{1, 2}, {3, 6}};
In[ ]: B = {{5, 6}, {7, 8}};
In[ ]: F = {{1, 2}, {3, 4}, {5, 6}};
In[ ]: G = {{1, 2, 3}, {4, 5, 6}}, {7, 8, 9}};

Then,

1. In[ ]: A + B

Out[ ]:

(
6 8
10 14

)
Addition, here, is defined since A and B have the same size. If we add the matrix A to
G then the result is undefined since they are different in size.

In[ ]: A + G

Out[ ]: Objects of unequal length and cannot be combined

2. In[ ]: 3 B - 2 A

Out[ ]:

(
13 14
15 12

)

3. In[ ]: A B // MatrixForm

Out[ ]:

(
5 12
21 48

)
Be careful ! This command evaluates the product as element by element product, not
the ordinary (dot) product you have learned in Linear Algebra class. To perform the
ordinary product, we use the command Dot as follows.

In[ ]: Dot[A, B] // MatrixForm

Out[ ]:

(
19 22
57 66

)
Note that the dot product above is defined since the number of columns of the first
matrix A equals the number of rows in the second matrix B. This case is not satisfied
when try to evaluate A · F for example.

In[ ]: Dot[A, F] // MatrixForm

Out[ ]: {{1,2},{3,6}} and {{1,2},{3,4},{5,6}} have incompatible shapes.

What if you like to find A5? If we use the command A^5, this will rise each element of
A to the power 5 as follows.

In[ ]: A^5

Out[ ]:

(
1 32

243 7776

)

The command MatrixPower[A,n] gives the nth matrix power of the matrix A by the
mean of ordinary product.

In[ ]: MatrixPower[A, 5]

Out[ ]:

(
2401 4802
7203 14406

)

4. The determinant is a useful value that can be computed from the elements of a square
matrix. It can be computed in Mathematica with the function Det. For example,

In[ ]: Det /@ {A, B, G}
Out[ ]: {0,−2, 0}

5. An n× n square matrix A is called invertible (also nonsingular) if there exists an n× n

square matrix B such that AB = BA = In where In denotes the n× n identity matrix
and the multiplication used is ordinary matrix multiplication. If this is the case, then
the matrix B is uniquely determined by A and is called the inverse of A, denoted by
A−1. The inverse can be computed in Mathematica with the function Inverse. For
example,

In[ ]: Inverse[B] // MatrixForm

Out[ ]:

(
−4 3
7
2 −5

2

)
A square matrix that is not invertible is called singular . A square matrix is singular if
and only if its determinant is 0. We have seen that Det[G] = 0, so that G is singular
matrix.

In[ ]: Inverse[G]

Out[ ]: Matrix {{1,2,3},{4,5,6},{7,8,9}} is singular

6. The command Eigenvalues[C] returns the eigenvalues of the square matrix C, while
Eigenvectors[C] returns its eigenvectors. The command Eigensystem[C] provides
both in a single call. To find the characteristic polynomial of C with respect to x, use
CharacteristicPolynomial[C, x], which involves the determinant and trace of C.

In[ ]: Eigenvalues[B]

Out[ ]:

{
1

2

(√
177+ 13

)
,
1

2

(
13−

√
177
)}

In[ ]: Eigenvectors[B]

Out[ ]:

(
1
14

(√
177− 3

)
1

1
14

(
−
√
177− 3

)
1

)
In[ ]: CharacteristicPolynomial[B, x]

Out[ ]: x2 − 13x− 2

4.11 Exercises

1. Produce the following lists:

(a) The square of the reciprocal of the first 12 integers

(b) The tuples that contains n, n2, n3 from n = 1 to n = 6

2. Sort the list in exercise (1a) in ascending order.

3. Find the total of the elements in the list in exercise (1a), and write your answer in
numerical form using 50 decimal digits.
Ans. 1.5649766384209024901665594306286946979587672228365

4. Find the value of sin x when x =
{
π,

π

2
,
π

3
,
π

4
,
π

5
,
π

6

}
.

Ans.

{
0, 1,

√
3
2
, 1√

2
,

√
5
8
−

√
5
8
, 1
2

}
5. For the list { {1,2}, {3,4}, {5,6}, {7,8}, {9,10} }

(a) reverse its elements,

(b) reverse the interior sub-lists only,

(c) reverse the list and each of its interior elements.

6. Which of the following functions is listable?

(a) Length

(b) Total

(c) Piecewise functions

7. How to generate a list such as follows using Mathematica?

{{1},{1,2},{1,2,3},{1,2,3,4},{1,2,3,4,5}}

8. Generate a list, called mylist, of 30 integers between −5 and 5 using seed va1ue
equals 372 by typing the following command on the same line:

SeedRandom[372]; mylist = RandomInteger[{-5, 5}, 30]

� Count the number of 0’s.
Ans. 4

� Find a way to count all those elements of the list which are not 1’s. Read more
about Except.
Ans. 27

9. Consider the 5× 5 tridiagonal matrix A =


2 1 0 0 0
1 2 1 0 0
0 1 2 1 0
0 0 1 2 1
0 0 0 1 2


(a) Use ToeplitzMatrix to write a code that generates such matrix.

(b) Find the determinant of the matrix A.
Ans. 6

(c) Find the matrix A−2

Ans.


55/36 −20/9 9/4 −16/9 35/36
−20/9 34/9 −4 29/9 −16/9
9/4 −4 19

4 −4 9/4
−16/9 29/9 −4 34/9 −20/9
35/36 −16/9 9/4 −20/9 55/36


(d) A symmetric matrix is a square matrix that is equal to its transpose. Show that

A is symmetric.

10. The Hilbert matrix is a square matrix whose element aij is
1

i+ j− 1
. Construct the

Hilbert matrix of order 5.

11. Find a unit vector having the same direction as v = (1,−2, 2,−3).
Hint: see the command Normalize.
Ans.

(
1/3

√
2,−

√
2/3,

√
2/3,−1/

√
2
)

12. The vector projection of a vector a on (or onto) a nonzero vector b (also known as
the vector component of a in the direction of b) is the orthogonal projection of a onto

a straight line parallel to b. It is a vector parallel to b, defined as a · b

|b|
where the

operator · denotes a dot product, and |b| is the length of |b|. Find the projection of
the vector a = (1, 2, 3) onto the vector b = (1,−2, 5).
Hint: see the command Projection.
Ans. (2/5,−4/5, 2)

13. Consider the following data:

data = {13,15,15,8,16,20,28,19,18,15,21,23,30,17,10,16,15,16,20,15}

Use Mathematica documentation to find for data the: Mean, Median, Variance,
and StandardDeviation.
Ans. 35/2, 16, 529/19, 23/

√
19

14. Consider the following data: data={{3,4},{6,6},{-1,5},{2,9},{4,0},{1,3}}.
Find the regression line equation that best fits the data.
Hint: see the command Fit

Ans. 4.79661− 0.118644x

5 Logic and Set Theory

5.1 Being Logical

In Mathematical logic, statements can have a value of True or False. These are called
Boolean expressions. This helps us to “make a decision” and write programs based on
the value of a statement. We have seen ==, which compares the left-hand side with the
right-hand side. For example

In[ ]: 3^2+4^2==5^2

Out[ ]: True

In[ ]: 3-4==1

Out[ ]: False

Other logical relations are

� x != y which is True if and only if x and y have different values.

� x < y which is True if and only if x is numerically less than y.

� x > y which is True if and only if x is numerically greater than y.

� x <= y which is True if and only if x is numerically less or equal to y.

� x >= y which is True if and only if x is numerically greater or equal to y.

For example,

In[ ]: 3^2+4^2>=5^2

Out[ ]: True

In[ ]: Sqrt[49]!=7

Out[ ]: False

In[ ]: E > Pi

Out[ ]: False

Note In logical expressions, ! means the negation (∼). If we type in Mathematica 3! this
means the factorial of 3 which equals 6. We can express the factorial function of an expr

using the Mathematica built-in function Factorial[expr].

In[ ]: 310! != Factorial[310]

Out[ ]: False

5.2 Truth Tables

One can combine logical statements with the usual Boolean operations And (∧), Or (∨) or
the equivalent shortcuts &&, ||. In general, for two statements A and B, the statement A ∧ B

is false if one of A or B is false and A ∨ B is true if one of them is true. In order to produce
all possible combinations of true and false cases, we use the command BooleanTable as
the following example shows.

In[ ]: BooleanTable[{p, q, p&&q}] /.{True->T, False->F} //Grid

Out[ ]:

T T T
T F F
F T F
F F F

In[ ]: BooleanTable[{p, q, p||q}] /.{True->T, False->F} //Grid

Out[ ]:

T T T
T F T
F T T
F F F

In logic, a tautology is a formula that is true in every possible interpretation. TautologyQ[bf]
gives True if all combinations of values of variables make the Boolean function bf yield True.

In[ ]: BooleanTable[(a||b) || (!a && !b)]

Out[ ]: {True, True, True, True}
In[ ]: TautologyQ[(a||b) || (!a && !b)]

Out[ ]: True

5.3 Element “∈”

In Mathematica, for a variable, one can specify certain domains. This means that the
variable takes its values from a specific type of data. The domains available are Algebraics,
Booleans, Complexes, Integers, Primes, Rationals and Reals. One of the fundamental
theorems in number theory is to show that π is not a rational number, i.e., is not of the form
m/n, where m and n ̸= 0 are integers. Look at the following examples:

In[ ]: Element[Pi, Rationals]

Out[ ]: False

In[ ]: Element[Sqrt[25], Integers]

Out[ ]: True

In[ ]: IntegerQ[Sqrt[25]]

Out[ ]: True

In[ ]: Element[131, Primes]

Out[ ]: True

In[ ]: PrimeQ[131]

Out[ ]: True

5.4 Handling Sets

Now it has been agreed among mathematicians that any mathematics starts by considering
sets, i.e., collections of objects. As we mentioned, the difference between Mathematical sets
and lists in Mathematica is that lists respect order and repetition, which is to say one can
have several copies of one object in a list. Sets are not sensitive about repeated objects,
e.g., the set {a, b} is the same as the set {a, b, b, a}. There is no concept of sets
in Mathematica and if necessary one considers a list as a set. If one wants to get rid of
duplication in a list, one can use

In[ ]: DeleteDuplicates[{a, b, b, a}]
Out[ ]: {a, b}

Considering two sets, the natural operations between them are union and intersection. Math-
ematica provides Union to collect all elements from different lists into one list (after removing
all the duplication) and Intersection for collecting common elements (again discarding
repeated elements). The following examples show how these commands work.

In[ ]: U = {1, 2, 3, 4, 5, 7};
In[ ]: A = {1, 4, 7, 3};
In[ ]: B = {5, 4, 3, 2}
In[ ]: Union[A,B]

Out[ ]: {1, 2, 3, 4, 5, 7}
In[ ]: Intersection[A,B]

Out[ ]: {3, 4}
In[ ]: Complement[U,A]

Out[ ]: {2, 5}

The command Complement[U,A] will give the elements of the universal set U which are not
in the set A.

In mathematics, the power set (or powerset) of any set S, written P(S), is the set of
all subsets of S, including the empty set and S itself. For example, the powerset of the set
{1, 3, 2} is

In[ ]: Subsets[{1, 3, 2}]
Out[ ]: {{}, {1}, {3}, {2}, {1, 3}, {1, 2}, {3, 2}, {1, 3, 2}}

In mathematics, two sets are said to be disjoint if they have no element in common. For
example,

In[ ]: DisjointQ[{2, 4, 6}, {1, 3, 5}]
Out[ ]: True

5.5 Quantifiers

In a statement like x4 + x2 > 0, Mathematica treats the variable x as having a definite,
though unspecified, value. Sometimes, however, it is useful to be able to make statements
about whole collections of possible values for x. You can do this using quantifiers ∀ ForAll

and ∃ Exists. In most cases, the quantifiers can be simplified by the Resolve function. The
command Reslove[expr ] attempts to resolve expr into a form that eliminates ForAll

and Exists quantifiers. For example, the following is true.

In[ ]: Resolve@Exists[x, x^2 + x^4 > 0]

Out[ ]: True

But, the following is false since x = 0 does not satisfy the statement.

In[ ]: Resolve@ForAll[x, x^2 + x^4 > 0]

Out[ ]: False

Example 5.1. Decide whether the statement is True or False.

1. There exists a natural number n such that n2 > 2n.

In[ ]: Resolve[Exists[n, n^2 > 2^n]]

Out[ ]: True

2. For all natural numbers n, if n ≥ 5, then n2 < 2n.

In[ ]: Resolve[ForAll[n, n>=5, n^2 < 2^n], Integers]

Out[ ]: True

3. The equation x2 + x = 1 has a negative real number solution.

In[ ]: Resolve[Exists[x, x<0, x^2+x==1], Reals]

Out[ ]: True

4. For all x ∈ R and x < 1, x3 − x ≤ 0.

In[ ]: Resolve[ForAll[x, x<1, x^3-x<=0], Reals]

Out[ ]: False

5. There exists boolean statements p and q such that (p ∨ q)∧ ∼ q is true.

In[ ]: Resolve[Exists[{p,q}, (p||q) && !q], Booleans]

Out[ ]: True

6. Every integer is either even or odd.
∀n ∈ Z,∃m ∈ Z such that n = 2m or n = 2m+ 1

In[ ]: Resolve[ForAll[n, Exists[m, n==2 m || n==2m+1]], Integers]

Out[ ]: True

5.6 Exercises

1. Construct the truth table for the statement p ∨ (q ∧ r).

2. Show that the statement (p ∧ q) ∨ (∼ p∨ ∼ q) is a tautology.

3. A contradiction is a formula that is false in every possible interpretation. Show that
the statement (p∧ ∼ q) ⇔ (p ⇒ q) is a contradiction.
Hint: use Mathematica’s help to read about the commands Equivalent (⇔) and
Implies (⇒).

4. Let U =
{
1, 2, 3, 4, 5, 6, 7

}
, A =

{
2, 5, 7

}
, and B =

{
3, 4, 6, 7

}
.

(a) Determine whether A, and the complement of B are disjoint or not.

(b) Evaluate A ∪ B, B ∩ A, and P(A) ∩ P(B).

(c) Show that A ⊆ A ∪ B and A ∩ B ⊆ A.

5. Let F be the set that contains 10 elements. Read more about the command Subsets

to answer the following questions:

(a) How many subsets of F are there contains at most 7 elements?
Ans. 968

(b) How many subsets of F are there contains at exactly 7 elements?
Ans. 120

(c) How many subsets of F are there contains between 3 and 7 elements?
Ans. 912

6. Which of the following are true?

(a) ∃x ∈ R such that 3x = x2.

(b) ∃x ∈ R such that 3x = x.

(c) ∀x ∈ R, x2 + 4x+ 5 ≥ 0.

(d) ∀x ∈ N, ∃y ∈ N such that x = 3y or x = 3y + 1 or x = 3y + 2.

(e) −1

2
≤ (x+ y)(1− xy)

(1 + x2) (1 + y2)
≤ 1

2
for all x, y ∈ R.

6 Number Theory

6.1 Primes

A prime number (or a prime) is a natural number greater than 1 that has no positive divisors
other than 1 and itself. A natural number greater than 1 that is not a prime number is called
a composite number. For example, 5 is prime because 1 and 5 are its only positive integer
factors, whereas 6 is composite because it has the divisors 2 and 3 in addition to 1 and 6.
Either the built-in function PrimeQ or CompositeQ will help you to determine whether an
integer is prime or composite.

� PrimeQ[expr] yields True if expr is a prime number, and yields False otherwise.

� CompositeQ[expr] yields False if expr is a prime number, and yields True otherwise.

For example, the number 13 is prime while 10100 + 1 is composite since

In[ ]: PrimeQ[13]

Out[ ]: True

In[ ]: CompositeQ[13]

Out[ ]: False

In[ ]: PrimeQ[10^100+1]

Out[ ]: False

In[ ]: CompositeQ[10^100+1]

Out[ ]: True

Example 6.1. Prove that for every integer n > 2, the number 22
n−1 + 1 is not a prime

number.

In[ ]: Resolve[ForAll[n, n>2, !PrimeQ[2^(2^n-1)+1]], Integers]

Out[ ]: True

The first few primes are 2, 3, 5, 7, 11, 13, · · ·. The first prime is 2, the second is 3, the sixth
is 13, and so on. In Mathematica, Prime[n] gives the nth prime number. For example,

In[ ]: Prime[1]

Out[ ]: 2

In[ ]: Prime[6]

Out[ ]: 13

In[ ]: Prime[2016]

Out[ ]: 17519

Prime[n] is listable function, that is it can find the primes at many orders or positions. For
example,

In[ ]: Prime[{1,6,2016}]
Out[ ]: {2, 13, 17519}

Example 6.2. Find the first 10 primes.

In[ ]: Prime[Range[10]]

Out[ ]: {2, 3, 5, 7, 11, 13, 17, 19, 23, 29}

To find a prime number above a certain value n, you may use NextPrime[n]. If you would
the kth prime above n you should use NextPrime[n,k]. For example,

In[ ]: NextPrime[4]

Out[ ]: 5

In[ ]: NextPrime[4,3]

Out[ ]: 11

If k is negative in NextPrime[n,k], then the result is the kth prime previous to n.

In[ ]: NextPrime[88,-5]

Out[ ]: 67

The number of primes less than or equal to x is evaluated using PrimePi[x]. For example,
there are 25 primes less that or equal to 100 since

In[ ]: PrimePi[100]

Out[ ]: 25

Example 6.3. How many 3−digit prime numbers are there?

In[ ]: PrimePi[999] - PrimePi[100]

Out[ ]: 143

6.2 Integer Factorization

In number theory, the prime factors of a positive integer are the prime numbers that divide
that integer exactly. The prime factorization of a positive integer is a list of the integer’s
prime factors, together with their multiplicities; the process of determining these factors is
called integer factorization. The fundamental theorem of arithmetic says that every positive
integer has a single unique prime factorization. To shorten prime factorizations, factors are
often expressed in powers (multiplicities). For example, 360 = 23 × 32 × 5. In Mathematica,
we use

In[ ]: FactorInteger[360]

Out[ ]: {{2, 3}, {3, 2}, {5, 1}}

The list of all integers that divides a number n can be evaluated using Divisors[n]. For
example,

In[ ]: Divisors[10]

Out[ ]: {1, 2, 5, 10}
In[ ]: Divisors[13]

Out[ ]: {1, 13}

Example 6.4. How many divisors are there for 13!.

In[ ]: Length@Divisors[13!]

Out[ ]: 1584

Example 6.5. How many integers are there divide both 1545 and 1230.

In[ ]: Intersection[Divisors[1545], Divisors[1230]] // Length

Out[ ]: 4

Sometimes, you need to know the highest power k of b such that bk divides n. To do this,
use the Mathematica function IntegerExponent[n,b]. For example, the highest power of
2 that divides 360 is 3.

In[ ]: IntegerExponent[360,2]

Out[ ]: 3

Example 6.6. With how many zeros will 130! end ?

In[ ]: IntegerExponent[130!,10]

Out[ ]: 32

The command IntegerLength[n] in Mathematica gives the number of digits in the base
10 representation of the integer n. Also, DigitCount[n,b] gives a list of the numbers of
1, 2, ..., b-1, 0 digits in the base-b representation of n.

Example 6.7. How many digits are there for 130! ?

In[ ]: IntegerLength[130!]

Out[ ]: 220

and they are distributed as follows:

14 1’s, 21 2’s, 14 3’s, 23 4’s, 24 5’s,
16 6’s, 20 7’s, 15 8’s, 22 9’s, 51 0’s,

since

In[ ]: DigitCount[130!]

Out[ ]: {14, 21, 14, 23, 24, 16, 20, 15, 22, 51}

Example 6.8. Find only the number of 0’s, 5’s and 9’s do appear in 168371921.

In[ ]: DigitCount[16837^1921, 10, {0, 5, 9}]
Out[ ]: {822, 790, 821}

46



Feras Awad @2025 6 NUMBER THEORY

6.3 Digits in Numbers

In this section we learn how to convert between numbers and lists. To obtain a list of the
decimal digits in the integer n, you may use IntegerDigits[n]. To construct an integer
from the list of its decimal digits use FromDigits[list].

Example 6.9. Reverse the integer 365435296161.

In[ ]: FromDigits[Reverse[IntegerDigits[365435296161]]]

Out[ ]: 161692534563

In[ ]: IntegerReverse[365435296161]

Out[ ]: 161692534563

Example 6.10. Find the sum of the digits of 1980!.

In[ ]: Total@IntegerDigits[1980!]

Out[ ]: 23418

6.4 Fibonacci Sequence

As we have seen in Chapter 3, the sequence 1, 1, 2, 3, 5, 8, 13, 21, · · · is called Fibonacci
sequence. By definition, the first two numbers in the Fibonacci sequence are 1 and 1,
and each subsequent number is the sum of the previous two. The Mathematica function
Fibonacci[n] gives the nth Fibonacci number.

In[ ]: Fibonacci[7]

Out[ ]: 13

Fibonacci[n] is listable function, so you can find, for example, the first 14 Fibonacci
numbers as follows.

In[ ]: Fibonacci[Range[14]]

Out[ ]: {1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233, 377}

Example 6.11. Find the sum of the first 100 Fibonacci numbers.

In[ ]: Total@Fibonacci[Range[100]]

Out[ ]: 927372692193078999175

Example 6.12. Determine whether the 13th Fibonacci number is prime.

In[ ]: PrimeQ[Fibonacci[13]]

Out[ ]: True

6.5 Number Theoretic Functions

In mathematics, the remainder is the amount “left over” after performing some computation.
In arithmetic, the remainder is the integer “left over” after dividing one integer by another
to produce an integer quotient (integer division). In computing, the modulo operation finds
the remainder after division of one number by another (sometimes called modulus).

Given two positive numbers, a (the dividend) and n (the divisor), a modulo n (abbreviated
as a mod n) is the remainder of the Euclidean division of a by n. For instance, the expression
“5 mod 2” would evaluate to 1 because 5 divided by 2 leaves a quotient of 2 and a remainder
of 1, while “9 mod 3” would evaluate to 0 because the division of 9 by 3 has a quotient of
3 and leaves a remainder of 0; there is nothing to subtract from 9 after multiplying 3 times
3.

Quotient and Remainder In Mathematica, if we divide a by n, then the quotient is
evaluated using Quotient[a, n] and the remainder (modulus) is Mod[a, n]. For example,

In[ ]: Quotient[5, 2]

Out[ ]: 2

In[ ]: Mod[5, 2]

Out[ ]: 1

In[ ]: QuotientRemainder[5, 2]

Out[ ]: {2, 1}
In[ ]: QuotientRemainder[9, 3]

Out[ ]: {3, 0}

If the remainder is 0 when we divide a by n, then we say that a is divisible by n. To test
whether m is divisible by n you may use Divisible[m,n] which yields True if m is divisible
by n, and yields False if it is not. For example,

In[ ]: Divisible[5, 2]

Out[ ]: False

In[ ]: Divisible[130!, 1210]

Out[ ]: True

In[ ]: Mod[130!, 1210]

Out[ ]: 0

GCD and LCM In mathematics, the greatest common divisor (gcd) of two or more inte-
gers, when at least one of them is not zero, is the largest positive integer that divides the
numbers without a remainder. For example, the GCD of 8 and 12 is 4.

In[ ]: GCD[8, 12]

Out[ ]: 4

The least common multiple of two or more integers is the smallest positive integer that is
divisible by all of them. Since division of integers by zero is undefined, this definition has
meaning only if the integers are different from zero. For example, the LCM of 8 and 12 is
24.

In[ ]: LCM[8, 12]

Out[ ]: 24

Example 6.13. Find the GCD and LCM for the numbers 2145, 1716, 9918.

In[ ]: GCD[2145, 1716, 9918]

Out[ ]: 3

In[ ]: LCM[2145, 1716, 9918]

Out[ ]: 14182740

In number theory, two integers a and b are said to be relatively prime, mutually prime, or
coprime if the only positive integer that evenly divides both of them is 1. That is, the only
common positive factor of the two numbers is 1. This is equivalent to their greatest common
divisor being 1. For example, 8, 9, and 11 are relatively prime since

In[ ]: CoprimeQ[8, 9, 11]

Out[ ]: True

The function CoprimeQ[n1, n2, ...] yields True if all pairs of the ni are relatively prime,
and yields False otherwise. For example, 6, 9, and 11 are not relatively prime since 3 divides
6 and 9.

In[ ]: CoprimeQ[6, 9, 11]

Out[ ]: False

6.6 Exercises

1. Find the 12th prime number?
Ans. 37

2. Determine whether 25 − 1 is prime or composite? What is the prime number next to
25 − 1? What is the 5th prime number next to 25 − 1? What is the 5th prime number
previous to 25 − 1?
Ans. Prime, 37, 53, 13

3. How many primes are there less than 1000? How many 6-digit primes are there?
Ans. 168, 68906

4. Factor the integer 12345. Write all its divisors. How many are there?
Ans. 3× 5× 823, {1, 3, 5, 15, 823, 2469, 4115, 12345}, 8

5. Find the least common multiple and the greatest common divisor of the integers 16,
24, 524.
Ans. 4, 6288

6. Find the quotient and reminder when 13! is divided by 2256.
Ans. 2760204, 576

7. How many positive integers are there divides both 12! and the 36th Fibonacci number,
but not 2100 − 1.
Ans. 18

8. Find the sum of the first 100 primes. Find the sum of the first 123 even integers.
Ans. 24133, 15252

9. A number is perfect if it is equal to the sum of its proper divisors. For example,
6 = 1+ 2+ 3 is perfect, while 18 ̸= 1+ 2+ 3+ 6+ 9 is not. You may use the com-
mand PerfectNumberQ to test whether a number is perfect or not. For example,
PerfectNumberQ[6] returns True, while PerfectNumberQ[18] returns False.

(a) Which of the following numbers in perfect: 496, 8218, and 8589869056?.

(b) Use the command PerfectNumber to find the first 7 perfect numbers.

10. A partition of a positive integer n, also called an integer partition, is a way of
writing n as a sum of positive integers. Two sums that differ only in the order of their
summands are considered the same partition. For example, 4 can be partitioned in five
distinct ways:

4 = 4

= 3 + 1

= 2 + 2

= 2 + 1 + 1

= 1 + 1 + 1 + 1

Use the command IntegerPartitions to answer the following questions.

(a) Give a list of all possible ways to partition the integer 5 into smaller integers ?

(b) Partitions 5 into at most 3 integers ?

(c) Partitions 5 into exactly 3 integers ?

(d) Find all partitions of 5 that involve only 1 and 3 ?

(e) How many ways are there to make change for 156 JOD with the paper coins 1,
5, 10, 20, and 50 with exactly 10 paper coins?

11. In mathematics, a Mersenne prime is a prime number of the form Mn = 2n − 1 for
some integer n. In order for Mn to be prime, n must itself be prime. This is true since

for composite n with factors a and b, n = ab. Therefore, 2n − 1 can be written as
2ab − 1, which always has the factors (2a − 1) and (2b − 1). The first five Mersenne
primes are Mn = 3, 7, 31, 127, 8191 corresponding to n = 2, 3, 5, 7, 13.

(a) In Mathematica, the command MersennePrimeExponentQ[n] returns True if n
is a Mersenne prime exponent, and False otherwise. Using this command, find
all Mersenne prime exponents smaller than 10000.

(b) The command MersennePrimeExponent[n] gives the nth Mersenne prime ex-
ponent. What is the 40th Mersenne prime exponent? How many digits are there
in the 40th Mersenne prime number?
Ans. 20996011, 8

12. Goldbach’s conjecture is one of the oldest and best-known unsolved problems in
number theory and all of mathematics. It states:

Every even integer greater than 2 can be expressed as the sum of two primes.

The expression of a given even number as a sum of two primes is called a Goldbach
partition of that number. The following are examples of Goldbach partitions for some
even numbers:

6=3+ 3

8=3+ 5

10=3+ 7 = 5+ 5

Write a code using Mathematica to find all possible Goldbach partitions given any even
integer n > 2, then use it to find all Goldbach partitions for n = 430.
Hint: Is is useful to use the command IntegerPartitions.

13. How does the Apply command function in Mathematica, and how can it be utilized,
alongside other commands, to express the prime factorization of a number such as 12,
as 22 · 31?

7 Computer Algebra and Solving Equations

The Wolfram Language has a variety of commands for algebraic manipulation operations like
expansion and factoring of polynomials, addition of fractions with unlike denominators, and
collection of terms with like variables. These commands have names that describe exactly
what they do, making it intuitive for new users to find the right commands for their needs.

7.1 Working with Polynomials, Ratioal and Transcendental Functions

One of the abilities of Mathematica is to handle symbolic computations, i.e., Mathematica
can comfortably work with symbols. For example, the Expand command does exactly what
its name says it does:

In[ ]: Expand [(x-2)(x-3)(x+1)^2]

Out[ ]: x4 − 3x3 − 3x2 + 7x+ 6

Mathematica can also do the inverse of this task, namely factorize an expression:

In[ ]: Factor[x^2 + 2 x + 1]

Out[ ]: (1+ x)2

While expansion of an algebraic expression is a simple and routine procedure, the factorization
of algebraic expressions is often quite challenging. My favorite example is this one.

In[ ]: Factor[x^10 + x^5 + 1]

Out[ ]:
(
1+ x+ x2

) (
1− x+ x3 − x4 + x5 − x7 + x8

)
Factor only works completely if all the roots are rational.

In[ ]: Factor[x^2 - 2]

Out[ ]: −2+ x2

Factor will not use complex numbers in its answer x2 + 1 unless one of the coefficients is
complex, so it won’t write x2 + 1 = (x− i)(x+ i).

In[ ]: Factor[x^2 + 1]

Out[ ]: 1+ x2

Setting Factor[poly,Extension->All] will extend the domain of coefficients to include
any irrational or complex numbers.

In[ ]: Factor[x^2 + 2, Extension -> All]

Out[ ]:
(
x− i

√
2
)(

x+ i
√
2
)

In[ ]: Factor[2 + 2 Sqrt[2] x + x^2, Extension -> All]

Out[ ]:
(
x+

√
2
)2

Expand and Factor do a nice job even when you use numerical coefficients.

In[ ]: Expand[(x - 1.25) (2 x - 0.5)]

Out[ ]: 0.625− 3.x+ 2x2

In[ ]: Factor[0.625 - 3 x + 2 x^2]

Out[ ]: 2.(−1.25+ x)(−0.25+ x)

The Expand and Factor commands also work for polynomials with more than one variable.

In[ ]: Expand[(2 x - 5 y + 3 z)^2]

Out[ ]: 4x2 − 20xy+ 25y2 + 12xz− 30yz+ 9z2

In[ ]: Factor[4 x^2 - 20 x y + 25 y^2 + 12 x z - 30 y z + 9 z^2]

Out[ ]: (2x− 5y+ 3z)2

The Simplify command produces an expression that is the shortest to write out. More often
than not, this will agree with what you think of as “simplest”. For example, x2 − 2x+ 1 = (x− 1)2

is factored when you Simplify it:

In[ ]: Simplify[x^2 - 2 x + 1]

Out[ ]: (−1+ x)2

However, if we ask Mathematica to simplify x3 + 2x2 − 2x− 1,

In[ ]: Simplify[x^3 + 2 x^2 - 2 x - 1]

Out[ ]: −1− 2x+ 2x2 + x3

Mathematica returns the expression, even though x3 + 2x2 − 2x− 1 can be factored as
(x− 1)

(
x2 + 3x+ 1

)
. Mathematica thinks that writing this cubic polynomial directly with

4 terms is simpler than factoring it into two expressions with a total of 5 terms.

Rational Functions A rational function is an expression of the form
P(x)

Q(x)
, where P and Q

are polynomials of a single variable x. The following are three common algebraic operations
involving rational functions. Combining terms over a common denominator can be achieved

with the Together command. For example, to combine
2

3x+ 1
+

5x

x+ 2
:

In[ ]: Together[2/(3 x + 1) + 5 x/(x + 2)]

Out[ ]:
4+ 7x+ 15x2

(2+ x)(1+ 3x)

Splitting up a rational function into its partial fraction decomposition can be done with the

Apart command. For example, to decompose
11x2 − 17x

(x− 1)2(2x+ 1)

In[ ]: Apart[(11 x^2 - 17 x)/((x - 1)^2 (2 x + 1))]

Out[ ]: − 2

(−1+ x)2
+

3

−1+ x
+

5

1+ 2x

The Apart command also does long division. The quotient
x5 − 2x2 + 6x+ 1

x2 + x+ 1
is found with:

In[ ]: Apart[(x^5 - 2 x^2 + 6 x + 1)/(x^2 + x + 1)]

Out[ ]: −1− x2 + x3 +
2+ 7x

1+ x+ x2

Transcendental Functions The Simplify command has reasonable success working many
basic identities involving the trigonometric and hyperbolic functions, such as

In[ ]: Simplify[Sin[x]^2 + Cos[x]^2]

Out[ ]: 1

In[ ]: Simplify[Sin[x/2] Cos[x/2]]

Out[ ]:
Sin[x]

2

In[ ]: Simplify[Cosh[x]^2 - Sinh[x]^2]

Out[ ]: 1

The TrigExpand and TrigReduce commands provide alternate methods of working with
trigonometric functions, performing the jobs suggested by their names as follows.

TrigExpand[expr ] Splits up sums and integer multiples that
appear in arguments of trigonometric func-
tions, and then expands out products of
trigonometric functions into sums of powers,
using trigonometric identities when possible.

For example:

In[ ]: TrigExpand[Sin[2 x]]

Out[ ]: 2Cos[x]Sin[x]

In[ ]: TrigExpand[Cos[x + y]]

Out[ ]: Cos[x]Cos[y]− Sin[x]Sin[y]

In[ ]: TrigExpand[Sin[2 x] Cos[3 x]]

Out[ ]: −Sin[x]

2
+

5

2
Cos[x]4Sin[x]− 5Cos[x]2Sin[x]3 +

Sin[x]5

2

TrigReduce[expr ] Given a trigonometric polynomial,
TrigReduce typically yields a linear
expression involving trigonometric functions
with more complicated arguments.

For example:

In[ ]: TrigReduce[2 Cos[x]^2]

Out[ ]: 1+ Cos[2x]

In[ ]: TrigReduce[2 Sin[x] Cos[y]]

Out[ ]: Sin[x− y] + Sin[x+ y]

In[ ]: TrigReduce[Sin[2 x] Cos[3 x]]

Out[ ]:
1

2
(−Sin[x] + Sin[5x])

The FullSimplify Command You may (and we usually do) prefer to use FullSimplify
when working with transcendental functions. Think of this command as a “full-strength”
version of Simplify. It uses a wider variety of expression simplification possibilities than
Simplify (although it could take a long time with complicated expressions).

Example 7.1. Find the exact value of
√
2+

√
3−

√
2
√
6+ 5.

In[ ]: FullSimplify[Sqrt[2] + Sqrt[3] - Sqrt[5+2Sqrt[6]]]

Out[ ]: 0

7.2 Equations and Their Solutions

Many commands are available for equation solving, from solving over specific domains to
returning general solutions to finding roots of equations. For new users, becoming familiar
with just a few of the most common commands is sufficient to solve many different types of
problems.

Solving equations and finding roots for different types of equations and relations are
one of the main endeavors of mathematics. For polynomials with one variable of the form
anx

n + an−1x
n−1 + · · ·+ a1x+ a0, it has been proved that there is no formula for finding

the roots when n ≥ 5 (in fact, when n = 3 or n = 4, the formulas are not that pretty!). This
forces us to find numerical ways to estimate the roots of the equations.

Using Wolfram Mathematica we have several commands at our disposal. There are
different kinds of equations and they require different commands to find the roots.

The Solve Command for an Equation Mathematica’s Solve command will solve an
equation for an unknown variable. You use it in the form:

Solve[ an equation , variable(s) to solve for ]

For example,

In[ ]: Solve[2 x + 5 == 9, x]

Out[ ]: {{x → 2}}

Notice that:

� Equations in Mathematica are written using the double-equal sign “==”.

� The output of Solve will always have an outer layer of curly braces { }.

� Inside the outer layer of curly braces, you’ll see substitution rules with the familiar “→”
syntax.

You can check that x=2 is the correct solution to the above equation:

In[ ]: 2 x + 5 == 9 /. {x -> 2}
Out[ ]: True

This means that after the substitution x=2, it is True that the left-hand side of the equation
equals the right-hand side.

Here are a few more examples involving the Solve command.

Example 7.2. Equations can have more than one solution. We can see numerical approxi-
mations of the solutions using the N command.

In[ ]: Solve[x^2 - 3 x + 1 == 0, x]

Out[ ]:

{{
x → 1

2
(3−

√
5)

}
,

{
x → 1

2
(3+

√
5))

}}
In[ ]: N@Solve[x^2 - 3 x + 1 == 0, x]

Out[ ]: {{x → 0.381966}, {x → 2.61803}}

Example 7.3. Here, two of the solutions are complex, with i standing for the familiar
√
−1.

In[ ]: Solve[x^3 + x^2 == -3 x, x]

Out[ ]:

{
{x → 0} ,

{
x → 1

2
(−1− i

√
11)

}
,

{
x → 1

2
(−1+ i

√
11)

}}
Example 7.4. If the equation involves other variables, Mathematica will treat them as
constants (or parameters, depending on your use).

In[ ]: Solve[y^2 - a*y == 2 a, y]

Out[ ]:

{{
y → 1

2
(a−

√
a
√
8+ a)

}
,

{
y → 1

2
(a+

√
a
√
8+ a)

}}
Note The Solve command works very well for equations involving polynomials. However,
it doesn’t have much success with trigonometric, exponential, logarithmic, or hyperbolic
functions.

The Solve Command for a System of Equations Solve can also be used to solve a
system of equations. In general, it is used in this form:

Solve[{ eqn 1, eqn 2, ...} , { var 1, var 2, ...}]

Example 7.5. Solve the system

{
3x+ 8y = 5

5x+ 2y = 7
.

In[ ]: Solve[{3 x + 8 y == 5, 5 x + 2 y == 7}, {x, y}]

Out[ ]:

{{
x → 23

17
, y → 2

17

}}

Example 7.6. Solve the system

{
2xy + y2 = −4

x+ y = 2
.

In[ ]: Solve[{2 x y + y^2 == -4, x + y == 2}, {x, y}]

Out[ ]:
{
{x → 2

√
2, y → 2(1−

√
2)}, {x → −2

√
2, y → 2(1+

√
2)}
}

Two sets of solutions are obtained.

Example 7.7. Solve the system

{
x+ y = 0

x+ y = 1

In[ ]: Solve[{x + y == 0, x + y == 1}, {x, y}]
Out[ ]: {}

There are no solutions to this system of equations.

Example 7.8. Solve the system

{
2x− y + z = 2

x+ y + z = 3

In[ ]: Solve[{2 x - y + z == 2, x + y + z == 3}, {x, y, z}]

Out[ ]:

{{
y → 1

2
+

x

2
, z → 5

2
− 3x

2

}}
The above command produced the warning message Solve::svars:

Equations may not give solutions for all solve variables. >> It alerts
you to the fact that you supplied only two equations, so one of the three variables
(here, it’s x) will be treated as a constant (or parameter) in the solution.

Equations and Numerical Solutions There are some equations, however, for which it is
mathematically impossible to find explicit formulas for the solutions. The Wolfram Language
uses Root objects to represent the solutions in this case.

In[ ]: Solve[2 - 4 x + x^5 == 0, x]

Out[ ]:

Even though you cannot get explicit formulas, you can still evaluate the solutions numerically.

In[ ]: Solve[2 - 4 x + x^5 == 0, x] // N

Out[ ]: {{x → −1.51851}, {x → 0.508499}, {x → 1.2436},
{x → −0.116792− 1.43845I}, {x → −0.116792+ 1.43845I}}

Or, you can use NSolve. The NSolve command uses efficient numerical techniques to
approximate roots of polynomials and a few other simple functions. It has the same syntax
as Solve.

In[ ]: NSolve[2 - 4 x + x^5 == 0, x]

Out[ ]: {{x → −1.51851}, {x → 0.508499}, {x → 1.2436},
{x → −0.116792− 1.43845I}, {x → −0.116792+ 1.43845I}}

NSolve doesn’t work at all with exponential, logarithmic, trigonometric, or hyperbolic func-
tions. You should solve these types of equations using FindRoot, and giving a starting value
for x. FindRoot has the form:

FindRoot[equation , {variable , estimation of a sol.}]

The estimate of a solution is just that a value that you think is close to or somehow near the
point where you expect to find a solution. FindRoot then iteratively searches for a solution
starting from that estimate.

Example 7.9. Find a root of ex + sinx near x = 0.

In[ ]: FindRoot[Sin[x] + Exp[x], {x, 0}]
Out[ ]:

{
x → −0.588533

}
Note that FindRoot[lhs == rhs,{x,x0,x1}] searches for a solution using x0 and x1 as
the first two values of x, avoiding the use of derivatives. If you specify only one starting value
of x, FindRoot searches for a solution using Newton methods. If you specify two starting
values, FindRoot uses a variant of the secant method.

Example 7.10. Find the positive root of cosx = x2 using both Newton and Secant methods.

In[ ]: FindRoot[Cos[x] == x^2, {x, 3}]
Out[ ]:

{
x → 0.824132

}
In[ ]: FindRoot[Cos[x] == x^2, {x, 3, 5}]
Out[ ]:

{
x → 0.824132

}
Also, FindRoot[lhs == rhs,{x,xstart,xmin,xmax}] searches for a solution, stopping the
search if x ever gets outside the range xmin to xmax.

Example 7.11. The equation tan x = 8− 17x2 has two real roots in the interval
[
− 1, 1

]
.

Find these two roots.

In[ ]: FindRoot[Tan[x] == 8 - 17 x^2, {x, 0.5, -1, 1}]
Out[ ]:

{
x → 0.652415

}
In[ ]: FindRoot[Tan[x] == 8 - 17 x^2, {x, -0.75, -1, 1}]
Out[ ]:

{
x → −0.722824

}
To solve a system of two or more equations in the unknowns x, y, etc., we use x = x0,
y = y0, · · · as initial estimates of a solution.

Example 7.12. Solve the system

{
− x3 + y2 = 5

3 cos x− x+ y = 4

The above equations have a simultaneous solution that’s close to x = 1 and y = 2. We can
pinpoint it with:

In[ ]: FindRoot[{y^2-x^3==5, y-x+3Cos[x]==4}, {x,1}, {y,1}]
Out[ ]: {x → 0.663687, y → 2.30051}

The Command Reduce Mathematica can solve algebraic equations and some involving
functions like sinx = a. Such equations often have infinitely many solutions, differing by
multiples of 2π. By default, Solve returns one solution with a warning about others, while
Reduce provides more details.

Reduce[expr, vars ]

reduces the statement expr by solving equations or

inequalities for vars and eliminating quantifiers

For example,

In[ ]: Reduce[a x + b == 0, x]

Out[ ]: (b == 0 && a == 0) ||

(
a! = 0 && x == −b

a

)
In[ ]: Reduce[Sin[x] == 1, x]

Out[ ]: C[1] ∈ Integers && x ==
π

2
+ 2πC[1]

In[ ]: Reduce[x^4 - 1 == 0, x]

Out[ ]: x == −1 || x == −I || x == I || x == 1

7.3 Inequalities

Just as the equation x2 + 3x = 2 asserts that x2 + 3x is equal to 2, so also the inequality
x2 + 3x > 2 asserts that x2 + 3x is greater than 2. In Mathematica, Reduce works not only
on equations, but also on inequalities. For example, the following pair of inequalities reduces
to a single inequality.

In[ ]: Reduce[{0 < x < 2, 1 < x < 4}, x]

Out[ ]: 1 < x < 2

These inequalities can never simultaneously be satisfied.

In[ ]: Reduce[{x < 1, x > 3}, x]

Out[ ]: False

This inequality yields three distinct intervals.

In[ ]: Reduce[(x - 1) (x - 2) (x - 3) (x - 4) > 0, x]

Out[ ]: x < 1 || 2 < x < 3 || x > 4

The ends of the intervals are at roots and poles.

In[ ]: Reduce[1 < (x^2 + 3 x)/(x + 1) < 2, x]

Out[ ]: −1−
√
2 < x < −2 || − 1+

√
2 < x < 1

The following inequality allows only finitely many intervals.

In[ ]: Reduce[{Sin[x] > 0, 0 <= x <= 2 Pi}, x]

Out[ ]: 0 < x < π

7.4 Exercises

1. Expand the function (1+ x)2(1− x)3.
Ans. −x5 + x4 + 2x3 − 2x2 − x+ 1

2. Factorize the polynomials

(a) (1+ x)3 + (1− x)3

Ans. 2
(
3x2 + 1

)
(b) x2 + 5

Ans.
(
x− i

√
5
) (

x+ i
√
5
)

3. Write
1

x− 1
+

x

x2 + 1
as single fraction.

Ans.
2x2 − x+ 1

(x− 1) (x2 + 1)

4. Decompose
x

x2 + 6x+ 5
into sum of partial fractions.

Ans.
5

4(x+ 5)
− 1

4(x+ 1)

5. Show that:

(a)
(1+

√
5)10 − (1−

√
5)10

1024
√
5

= 55

(b) tan

(
3π

11

)
+ 4 sin

(
2π

11

)
=

√
11

(c)
etanh

−1 x − e− tanh−1 x

etanh
−1 x + e− tanh−1 x

= x

6. Find all the coefficients of (2+ 3x)4. Hint: see CoefficientList.
Ans. 16, 96, 216, 216, 81

7. What is the coefficient of xy3 when we expand (x+ y)4 ? What is the coefficient of
y2 ? Hint: see the command Coefficient.
Ans. 4, 6x2, 6x2

8. Find the exact value of the number
√

3 +
√
8−

√
2.

Ans. 1

9. Compare TrigExpand and TrigReduce on the expression sin2 x+ tan2 x.

10. Find the quotient and remainder of the polynomial division
x2 + x+ 1

2x+ 1
.

Hint: see PolynomialQuotient and PolynomialRemainder.

Ans.
x

2
+

1

4
,
3

4

11. Solve each of the following.

(a) x2 − 3x+ 1 = 0

(b) x5 − 3x+ 1 = 0

(c) x+ y+ z = 3 , 2x2 + 3y2 − 6z2 = −1 and 3z2 = 9x2 − 6y3

(d) tan3 x+ e2x = 10 over the interval
[
1, 5
]

Ans. 1.87452

(e) x2 ≥ 1

12. (a) Solve the equation
∣∣x− 1

∣∣ = 3 over real numbers.

(b) Find integer solution(s) for the equation
(
x2 − 3x+ 1

)x+1
= 1.

Ans. −1, 0, 1, 3

13. Solve the equation sin2 x− 3 sinx+ 2 = 0 for sinx

14. The Taxicab Number. The famous number theorist G. H. Hardy (1887-1947) was
visiting S. Ramanujan (1887-1920) when Ramanujan was ill in a hospital. Hardy
said that the number of his taxicab is 1729, seemed to him to be a dull number.
Ramanujan responded that, on the contrary, 1729 is the sum of two cubes in two
ways. Prove Ramanujan claim. Use the command Solve then use the new function
PowersRepresentations.
Ans. 1, 12 or 9, 10

15. The built-in function FindInstance[expr, vars] finds an instance of vars that
makes the statement expr be True. Use Mathematica’s help to read more about this
function, and then use it to solve exercise (14).

16. For what truth values of the statements a, b, c, and d will the following proposition
is true: (∼ a ∨ b) ∧ (c∧ ∼ d).
Hint: again, you may use FindInstance.
Ans. a b c d

T T T F
F T T F
F F T F

8 Single Variable Calculus

Two important machineries in calculus are differentiation and integration, and both use the
concept of limit. We assume the reader is familiar with calculus as all the material in these
notes.

8.1 Limits

Mathematica provides the command Limit for exploring the limit of a function. If f is a
function of a single variable, Mathematica evaluates lim

x→a
f(x) with the following syntax:

Limit[ function , variable -> a ]

For example, lim
x→0

sin x

x
= 1 as shown below.

In[ ]: Limit[Sin[x]/x, x -> 0

Out[ ]: 1

The limit of the function
|x|
x

when x → 0 does not exist.

In[ ]: Limit[RealAbs[x]/x, x -> 0]

Out[ ]: Indeterminate

Be careful! The function
|x|
x

has a different limiting value at x = 0, depending on whether

you approach from above or below. We know that the right-hand limit lim
x→0+

|x|
x

= 1 and

the left-hand limit lim
x→0−

|x|
x

= −1. Therefore lim
x→0

|x|
x

does not exist. To calculate a left-

hand lim
x→a−

f(x), you have to specify the Direction option in the Limit command. To find

lim
x→0−

|x|
x
, you use:

In[ ]: Limit[RealAbs[x]/x, x -> 0, Direction -> "FromBelow"]

Out[ ]: −1

Similarly, to find the right-hand limit lim
x→0+

|x|
x
, you can use:

In[ ]: Limit[RealAbs[x]/x, x -> 0, Direction -> "FromAbove"]

Out[ ]: 1

Note: To evaluate the limit from both real directions we use Direction → Reals or
Direction → "TwoSided".

The following examples show some sample limit computations. Mathematica can com-
pute most limits, even those that involve infinite limits, or limits at infinity:

63



Feras Awad @2025 8 SINGLE VARIABLE CALCULUS

1. lim
x→0

ex − x− 1

x

In[ ]: Limit[(Exp[x] - x - 1)/x, x -> 0]

Out[ ]: 0

2. lim
x→3

1− x

(x− 3)2

In[ ]: Limit[(1 - x)/(x - 3)^2, x -> 3]

Out[ ]: −∞

3. lim
x→∞

x√
x2 + 1

In[ ]: Limit[x/Sqrt[x^2 + 1], x -> Infinity]

Out[ ]: 1

4. lim
x→−∞

x√
x2 + 1

In[ ]: Limit[x/Sqrt[x^2 + 1], x -> -Infinity]

Out[ ]: −1

5. lim
x→0

3x− sin(3x)

4x− tan(4x)

In[ ]: Limit[(3 x - Sin[3 x])/(4 x - Tan[4 x]), x -> 0]

Out[ ]: − 27

128

6. lim
x→∞

sin x

In[ ]: Limit[Sin[x], x -> Infinity]

Out[ ]: Indeterminate

7. lim
x→1

ln x

1− x

In[ ]: Limit[Log[x]/(1 - x), x -> 1]

Out[ ]: −1

8.2 Differentiation

If f is a function of a single variable, Mathematica will understand the symbol f′ as the
derivative of f, where the prime is entered using the single-quote character. For example,

In[ ]: f[x ] := x^4

In[ ]: f′[x]

Out[ ]: 4x3

64



Feras Awad @2025 8 SINGLE VARIABLE CALCULUS

Higher-order derivatives follow with the usual notation:

In[ ]: {f′′[x],f′′′[a]}
Out[ ]: {12x2, 24a}

Also, you can use the differential operator D for computing derivatives.

D[ function , variable]

For example,

In[ ]: f[x ] := x^4

In[ ]: D[f[x],x]

Out[ ]: 4x3

To calculate a second derivative, you can use D either of these short-hand formats:

In[ ]: D[f[x],x,x]

Out[ ]: 12x2

In[ ]: D[f[x],{x,2}]
Out[ ]: 12x2

Similarly, for the third derivative of f we can use either of the following:

In[ ]: D[f[x],x,x,x]

Out[ ]: 24x

In[ ]: D[f[x],{x,3}]
Out[ ]: 24x

Example 8.1. If g(t) = t2 + t3 + ln t, find g′′(1).

In[ ]: g[t ] := t^2+t^3+Log[t]

In[ ]: g′′[1]

Out[ ]: 7

In[ ]: D[g[t],{t,2}] /. {t -> 1}
Out[ ]: 7

Example 8.2. Find the equation of the tangent line for the function g(t), given in the previ-
ous example, at t = 1. Remember that the equation of tangent is given by y = g(1) + g′(1)(t− 1).
So,

In[ ]: Simplify[g[1] + g′[1] (t - 1)]

Out[ ]: −4+ 6t

Example 8.3. Evaluate
d

dx

[
x sin (3πex)

]∣∣∣
x=ln 2

.

In[ ]: D[x Sin[3 Pi Exp[x]], x] /. x -> Log[2]

Out[ ]: 6πLog[2]

8.3 Maximum and Minimum

In Mathematica, Maximize[f,x] try to find the absolute maximum value of f with respect
to x on its natural domain or a given domain. While Minimize[f,x] try to find the absolute
minimum value of f with respect to x.

In[ ]: Maximize[-2 x^2 - 3 x + 5, x]

Out[ ]:

{
49

8
,

{
x → −3

4

}}
In[ ]: Maximize[{Sin[x], 0 <= x <= 2 Pi}, x]

Out[ ]:
{
1,
{
x → π

2

}}
In[ ]: Minimize[{Cos[x], -3 Pi/4 <= x <= 5 Pi/6}, x]

Out[ ]:

{
−
√
3

2
,

{
x → 5π

6

}}
Finding local maximum and minimum values of f is available in Mathematica using the
commands FindMinimum and FindMaximum as follows:

In[ ]: FindMinimum[x^3-x, x]

Out[ ]: {−0.3849, {x → 0.57735}}
In[ ]: FindMaximum[x^3-x, {x,-4}]
Out[ ]: {0.3849, {x → −0.57735}}

8.4 Integration

F(x) is an antiderivative of f(x) if F′(x) = f(x). The symbol

∫
f(x)dx means “find all

antiderivatives of f(x).” Because all antiderivatives of a given function differ by a constant,

we usually write

∫
f(x)dx = F(x) + C where C represents an arbitrary constant. You can

use the Integrate command to compute

∫
f(x)dx. It has the form:

Integrate[f[x], x]

You specify a function or an expression to integrate, as well as the variable in which the

integration is to take place. For example, to compute

∫
x3 dx, use this syntax:

In[ ]: Integrate[x^3, x]

Out[ ]:
x4

4

Mathematica can integrate almost every integral that can be done using standard integration
methods (e.g., substitution, integration by parts, partial fractions). Here are some typical
integrations:

1.

∫
x ln x dx

In[ ]: Integrate[x Log[x], x]

Out[ ]: −x2

4
+

1

2
x2Log[x]

2.

∫
1

1− x2
dx

In[ ]: Integrate[1/(1 - x^2), x]

Out[ ]: −1

2
Log[1− x] +

1

2
Log[1+ x]

3.

∫
e
√
x

√
x
dx

In[ ]: Integrate[Exp[Sqrt[x]]/Sqrt[x], x]

Out[ ]: 2e
√
x

4.

∫ √
4− 9x2 dx

In[ ]: Integrate[Sqrt[4 - 9 x^2], x]

Out[ ]:
1

2
x
√
4− 9x2 +

2

3
sin−1

(
3x

2

)
5.

∫
x

1+ x4
dx

In[ ]: Integrate[x/(1 + x^4), x]

Out[ ]:
tan−1

(
x2
)

2

6.

∫
xex

(1+ x)2
dx

In[ ]: Integrate[x Exp[x]/(1 + x)^2, x]

Out[ ]:
ex

1+ x

7.

∫
cos
(
sin
(
y2
))

dy

In[ ]: Integrate[Cos[Sin[y^2]], y]

Out[ ]:

∫
cos
(
sin
(
y2
))

dy

As you see in the last of the examples above, the integrand has no closed-form anti-derivative.
So, Mathematica will return your input unevaluated whenever it can’t handle an integral.

This can mean either that it’s not possible to find an antiderivative in closed form or that
Mathematica hasn’t yet been programmed to do the integral.

Definite Integral A definite integral

∫ b

a

f(x)dx is computed in Mathematica with this

form of the Integrate command:

Integrate[f[x], {x,a,b}]

Mathematica will try to find an antiderivative first, then evaluate it at the endpoints and
subtract (according to the Fundamental Theorem of Calculus). Here are some examples:

1.

∫ 2

−1

x2 dx

In[ ]: Integrate[x^2, {x, -1, 2}]
Out[ ]: 3

2.

∫ ∞

0

1

4+ y2
dy

In[ ]: Integrate[1/(4 + y^2), {y, 0, Infinity}]

Out[ ]:
π

4

3.

∫ 1

0

√
sin (t2) dt

In[ ]: Integrate[Sqrt[Sin[t^2]], {t, 0, 1}]

Out[ ]:

∫ 1

0

√
sin (t2)dt

In the last example, Mathematica can’t evaluate an antiderivative at the endpoints. The

NIntegrate command finds a numerical approximation for a definite integral

∫ b

a

f(x)dx

with the following syntax:

NIntegrate[f[x], {x,a,b}]

Notice that its syntax is the same as the Integrate command. For example,

In[ ]: NIntegrate[x^2, {x, -1, 2}]
Out[ ]: 3

In[ ]: NIntegrate[Sqrt[Sin[t^2]], {t, 0, 1}]
Out[ ]: 0.486177

The NIntegrate command doesn’t attempt to find a symbolic antiderivative, so it’s quick

and it works with almost all integrands, including

∫ 1

0

√
sin (t2)dt for which Integrate

failed (as you saw above).

8.5 Sequences

A sequence is a function whose domain is the set of positive integers. The terms of the
sequence

{
an
}
are a1, a2, a3, · · ·. The term an is called the general term of the sequence.

For example, the sequence
1

2
,
2

3
,
3

4
,
4

5
,
5

6
,
6

7
, · · · has the general term an =

n

n+ 1
. In Math-

ematica, we attempt to find a simple function that yields the sequence
{
an
}
when given

successive integer arguments by using the command FindSequenceFunction.

In[ ]: FindSequenceFunction[{1/2, 2/3, 3/4, 4/5}, n]

Out[ ]:
n

n+ 1

A sequence
{
an
}
is said to converge to the limit ℓ if lim

n→∞
an = ℓ. A sequence that does not

converge to some finite limit is said to diverge. The command DiscreteLimit gives the
limit lim

n→∞
fn for the sequence fn as n tends to infinity over the integers.

Example 8.4. Determine whether the sequence converges or diverges.

1.

√
9n2 + 4n− 7

2n+ 1

In[ ]: DiscreteLimit[Sqrt[9n^2+4n-7]/(2n+1), n -> Infinity]

Out[ ]:
3

2

2.
nn

n!

In[ ]: DiscreteLimit[n^n/n!, n -> Infinity]

Out[ ]: ∞

8.6 Series

You can use the Sum command to add up a finite number of terms of an indexed expression.

To find
b∑

n=a
for an expression, you type:

Sum[ expression , {n,a,b}]

For example, the value of
20∑
n=1

n2 is

In[ ]: Sum[ n^2 , {n,1,20}]
Out[ ]: 2870

The Sum command has moderate success even with certain symbolic summations. For

example,
k∑

n=0

rn is a partial sum for a geometric series:

In[ ]: Sum[ r^n , {n,0,k}]

Out[ ]:
−1+ r1+k

−1+ r

Example 8.5. Show that 1+ 2+ 3+ · · ·+ k =
k(k+ 1)

2
.

In[ ]: Sum[ n , {n,1,k}]

Out[ ]:
1

2
k(k+ 1)

If you want an approximate value for a summation, use NSum instead. It has the same syntax
as Sum. For example,

In[ ]: NSum[Sin[n], {n, 0, 500}]
Out[ ]: 1.4903

Both the Sum and NSum commands allow you to specify an infinite range, providing the
ability to evaluate certain infinite series. Each is also pretty good at recognizing series that
do not converge. In particular, Sum can symbolically reproduce almost all standard series
computations found in Calculus books and tables. For example,

1.
∞∑
n=0

(arn) (* Geometric Series *)

In[ ]: Sum[a r^n, {n, 0, Infinity}]

Out[ ]:
a

1− r

2.
∞∑
n=1

1

n
(* Divergent Series *)

In[ ]: Sum[1/n, {n, 1, Infinity}]
Out[ ]: Sum::div: Sum does not converge.

3.
∞∑
n=1

(−1)n

n
(* Alternating Series *)

In[ ]: Sum[(-1)^n/n, {n, 1, Infinity}]
Out[ ]: −Log[2]

4.
∞∑
n=1

1

n2
(* Convergent p-Series *)

In[ ]: Sum[1/n^2, {n, 1, Infinity}]

Out[ ]:
π2

6

5.
∞∑
n=0

2n

n!
(* Taylor Series of ex *)

In[ ]: Sum[2^n/n!, {n, 0, Infinity}]
Out[ ]: e2

6.
∞∑
n=0

sin (e−n) (* Unevaluated *)

In[ ]: Sum[Sin[Exp[-n]], {n, 0, Infinity}]

Out[ ]:

∞∑
n=0

sin
(
e−n
)

Notice that Sum returns its input unevaluated if it doesn’t have a known simplification or
result for it, or if it is known to not converge (see the last example above). If you do not
receive an error message telling you that a series does not converge then you can investigate
using NSum.

In[ ]: NSum[Sin[Exp[-n]], {n, 0, Infinity}]
Out[ ]: 1.41477

The command SumConvergence gives conditions for the sum
∞∑
n

fn to be convergent. For

example, the p−series
∞∑
n

1

np
converges for all real numbers p > 1.

In[ ]: Assuming[Element[p, Reals], SumConvergence[1/n^p, n]]

Out[ ]: p > 1

Example 8.6. Find the interval of convergence for the real power series
∞∑
n

xn

n 3n
.

In[ ]: SumConvergence[(x^n)/(n 3^n), n, Assumptions->Element[x,Reals]]

Out[ ]: −3 ≤ x < 3

8.7 Taylor Polynomials

Recall that if a function f satisfies certain reasonable conditions, then it can be approximated
by a polynomial pn(x) of degree n near a point x = a defined by:

pn(x) = f(a) +
f′(a)

1!
(x− a) +

f′′(a)

2!
(x− a)2 + · · ·+ f(n)(a)

n!
(x− a)n

A composition of the Normal and Series commands as follows will more easily produce the
Taylor polynomial of degree n for a function about a point x=a:

Normal[Series[ function, {x, a, n}]]

For example,

In[ ]: Normal[Series[Exp[x], {x, 0, 6}]]

Out[ ]: 1+ x+
x2

2
+

x3

6
+

x4

24
+

x5

120
+

x6

720

The following command will give an approximate value of e3 using the Taylor polynomial of
degree 6 for ex about x = 0:

In[ ]: Normal[Series[Exp[x], {x, 0, 6}]] /. {x -> 3} // N

Out[ ]: 19.4125

The Series command by itself actually gives a Taylor polynomial together with a remainder
term which shows its order using the notation O[x].

In[ ]: Series[Exp[x], {x, 0, 6}]

Out[ ]: 1+ x+
x2

2
+

x3

6
+

x4

24
+

x5

120
+

x6

720
+ O[x]7

Applying Normal to this result removes the remainder term and give the Taylor polynomial.

8.8 Exercises

1. Evaluate each of the following limits:

(a) lim
x→3

x2 − 9

x− 3
.

(b) lim
x→∞

3x− sin(3x)

4x− tan(4x)
.

(c) lim
x→∞

tan−1 x

ex
.

(d) lim
x→1

1− x∣∣x2 + x− 2
∣∣ .

(e) lim
x→∞

(
1+

a

x

)bx
2. Find the domain and range of each of the following functions.

Hint: use the commands FunctionDomain and FunctionRange

(a) f(x) =
1

x− x2
.

Ans. x < 0 ∨ 0 < x < 1 ∨ x > 1, y < 0 ∨ y ≥ 4

(b)
√∣∣x∣∣.
Ans. R, y ≥ 0

(c) f(x) = 2 ln x and g(x) = ln
(
x2
)
.

3. Let f(x) =

{
2x2 x < 1

−3x+ 5 x ≥ 1
. Find: f′(2) and f′(1).

Ans. −3, does not exist

4. Let f(x) = x3 + ex + sin x. Find: f′(x) and f′′(0).

5. Find
d87

dx87

[
x sin x

]
at x =

π

2
.

Ans. −87

6. Given that f(x) = x3 − 3x2 + 3 ; x ∈
[
− 2, 4

]
. Find for f :

(a) Critical numbers.

(b) Intervals of increasing and decreasing.

(c) Absolute maximum and absolute minimum values.

(d) Intervals of concavity.

(e) Inflection points.

7. Evaluate each of the following integrals:

(a)

∫
x2
√
1+ x2 dx

(b)

∫
x+ 1

x2 − 9
dx

(c)

∫
e1/x

x2
dx

(d)

∫
x2 cos x dx

(e)

∫ 4

1

x2 + 1√
x

dx

(f)

∫ ∞

0

e−x2 dx

(g)

∫ π/2

0

√
sin x dx

8. Find the values of A such that lim
x→A

(
1

x− 3
− 6

x2 − 9

)
= lim

x→A

x2 − x− 12

36x− 144
.

Ans. −9 or 3

9. Assume n ≥ 1, find a formula for the nth order derivative of each of the following
functions.

(a) f(x) = xex.

(b) g(x) = sinx

(c) h(x) = x cosx

10. Show that (1+ 2+ · · ·+ n)2 = 13 + 23 + · · ·+ n3.

11. Evaluate the following:

(a)
∞∑
n=0

(−1)n

(b)
∞∑
n=0

(−1)n x2n

(2n)!

12. Find Taylor polynomial of degree 10 for f(x) =
√
x about x = 1 and use it to approx-

imate the value of
√
1.5.

13. Determine whether the series converges or diverges.

1+
1

2
+

1

3
− 1

4
− 1

5
− 1

6
+

1

7
+

1

8
+

1

9
− · · ·

Ans.
1

9

(
2π

√
3+ log(8)

)
14. Find

∞∑
n=0

1

F2n
where Fm is the mth term of Fibonacci sequence.

Ans.
1

2

(
7−

√
5
)

15. Find the interval of convergence of the power series
∞∑
n

(x− 3)n√
n

.

16. Find the general term of each of the following sequences.

(a) −1

2
,
1

8
,− 1

24
,
1

64
,− 1

160
,

1

384
, · · ·

(b) 2, 3, 5, 7, 11, 13, · · ·
(c) 1, 0,−1, 0, 1, 0,−1, 0, · · ·

17. Determine whether the sequence −1

5
,
1

25
,− 1

125
,

1

625
, · · · converges or diverges.

9 Graphics in Mathematica

9.1 Making Graphs

There are many ways that you can plot a two-dimensional picture in Mathematica. In this
chapter, we will concentrate on drawing the graph of a function. You draw the graph of a
function y = f(x) with the Plot command. To see the graph of f(x) = x2 over the interval
[−3, 2], type

In[ ]: Plot[x^2, {x, -3, 2}]

Out[ ]:

In general, to plot a function of x over an interval x ∈ [a, b], you type:

Plot[f(x), {x, a, b}]

When you use the Plot command, you usually don not need to specify the vertical range of
the graph. Mathematica automatically adjusts the vertical range to show you the full range
of (vertical) values covered by the plot. If you wish, you can explicitly control the horizontal
and vertical range by adding a PlotRange specification in the Plot command. For example:

In[ ]: Plot[x^3, {x, -2, 2}, PlotRange -> {-3, 3}]

Out[ ]:

Here you see only the portion of the graph with y−values between −3 and 3. Thus, using
Plot with the PlotRange option is similar to setting up a view window on a graphing
calculator. You can specify a range of x−values as well, by using the option PlotRange->

{{Xmin, Xmax} , {Ymin, Ymax}}.

Proportion and Aspect Ratio When you are working in a Mathematica notebook, you
can resize a picture by first clicking on it and then dragging along one of its comers or edges
with the mouse. By default, no matter how big or small a picture you create, the aspect ratio

of the picture (the ratio of its height to width) will remain the same. The default aspect
ratio is very close to the relative proportions of a standard credit card. You can change the
aspect ratio for a Plot command by directly specifying a value for its AspectRatio. For
example, the height of the following picture is about one-fifth its width:

In[ ]: Plot[2 Sin[x], {x, -3, 3}, AspectRatio -> 1/5]

Out[ ]:

To make the units on the x− and y−axes have the same length, we set AspectRatio to
Automatic.

In[ ]: Plot[2 Sin[x], {x, -3, 3}, AspectRatio -> Automatic]

Out[ ]:

Plotting Multiple Expressions You can plot several functions in the same picture by
listing all the functions separated by commas, enclosing them in curly braces. For example:

In[ ]: Plot[{Sin[x], Cos[x]}, {x, -Pi, Pi}]

Out[ ]:

Several graphics produced independently can be combined into a single picture by using the
Show command. First, name each of the graphics as they are produced. Then, use the Show
command to combine them into a single output.

In[ ]: graph1 = Plot[Cos[x], {x, -3, 1}];
In[ ]: graph2 = Plot[Sin[x], {x, 0, 3}];
In[ ]: graph3 = Plot[0.5, {x, -2, 4}];
In[ ]: Show[graph1, graph2, graph3]

Out[ ]:

Notice that the combined graphic above is shown only over the range [−3, 1], which is the
range of the first graphic listed in the Show command. This is a default behavior of Show,
it arranges its output according to the options attached to the first graphic. We can change
this behavior by specifying PlotRange -> All in the Show command. The result is:

In[ ]: Show[graph1, graph2, graph3, PlotRange -> All]

Out[ ]:

We can use the GraphicsRow command to ask Mathematica to show several plots side by
side:

In[ ]: Show[GraphicsRow[{graph1, graph2}]]

Out[ ]:

The GraphicsRow command has several options that can be specified. For example, one
(Alignment)allows for adjusting the alignment of the graphics, while another (Dividers)specifies
whether lines should be drawn between and/or above and below the graphics. Mathematica
also has a GraphicsGrid command that allows you to arrange graphics in a two-dimensional
display.

The PlotStyle Option You can add more properties to a graph by adjusting the color,
thickness, opacity level, and dashing pattern of the plot. This can be done by specifying one
or multiple PlotStyle options for the Plot command. It has the syntax:

Plot[f[x], {x,a,b}, PlotStyle -> {Option 1, Option 2,...}]

For example, we will make the sine curve red with a dashed line and the cosine curve a thick
blue. Here is how we define these style features, using the names style1 and style2.

In[ ]: style1 = {Red, Dashing[{0.02, 0.02}]};
In[ ]: style2 = {Blue, Thickness[0.02]};
In[ ]: Plot[{Sin[x],Cos[x]},{x,-Pi,Pi}, PlotStyle->{style1,style2}]

Out[ ]:

Thickness[0.02] means that lines will be drawn to be about 2% of the width of the
graphic, while Dashing[{0.02,0 .04}] means that the curve will be drawn solid for about
2% of the width of the graphic, then 2% will be omitted, and so on.

9.2 Plotting Curves

In the previous section you saw how to use Plot to draw curves that are graphs of functions.
But not all curves are the graphs of functions.

Parametric Plots A two-dimensional (2D) parametric curve is usually defined by an or-
dered pair of coordinate functions (x(t), y(t)), with the parameter t allowed to vary over
some set. You use the ParametricPlot command to draw such a curve over a fixed interval
[a, b]. The command has the form:

ParametricPlot[{x[t],y[t]}, {t,a,b}]

For example, to see the curve
(
t2, t+ 1

)
for t ∈ [−2, 2], we use:

In[ ]: ParametricPlot[{t^2, t + 1}, {t, -2, 2}]

Out[ ]:

As with the Plot command, Mathematica will always, by default, show you a large enough
picture to see all values on the curve. If necessary, you can specify a given PlotRange

for ParametricPlot. Also, unlike the Plot command, ParametricPlot uses the option
AspectRatio → Automatic by default. Thus, curves which are circles will truly appear as
circles:

In[ ]: ParametricPlot[{Cos[t], Sin[t]}, {t, 0, 2 Pi}]

Out[ ]:

However, setting AspectRatio → Automatic is not always the right thing to do, as you
will see in drawing the curve

(
t, t3

)
where t ∈ [0, 5] (try it by your self). You can draw a

better picture with the following command

In[ ]: ParametricPlot[{t,t^3}, {t,0,5}, AspectRatio->1/GoldenRatio]

Out[ ]:

We suggest you redraw the graphic either by setting AspectRatio→l which gives a square
graphic, or by setting AspectRatio→l/GoldenRatio which gives the standard proportion
graphic you see with most other commands.

Multiple Curves The ParametricPlot command lets you plot several curves together,
just like the Plot command. To do this, you include the formulas for the curves in a list. For
example, to draw the horizontal line (t, 1), the vertical line (2, t), and the curve (cos t, t)
for t ∈

[
−π

2
, π
2

]
, we use the command

In[ ]: ParametricPlot[{{t,1}, {2,t}, {Cos[t],t}}, {t,-Pi,Pi}]

Out[ ]:

Plotting in Polar Coordinates If a curve is given in polar coordinates (r, θ), where r is
the radius, θ is the angle in radians, and r = f(θ) for θ ∈ [θ1, θ2], then we can draw it with
the PolarPlot command using the syntax:

PolarPlot[f[theta], {theta, θ1,θ2}]

Notice that this form is very similar to that of the Plot command you already know. For
example, to plot the three-leaf rose r = 2 cos(3θ), use:

In[ ]: PolarPlot[2 Cos[3 theta], {theta, 0, 2Pi}]

Out[ ]:

The default setting for PolarPlot is AspectRatio→Automatic. As we discussed in
the case of ParametricPlot, you may want to directly specify the value of the option
AspectRatio for a PolarPlot command that does not display well.

Just as with the Plot and ParametricPlot commands, you can plot more than one polar
curve at the same time by entering all of them using a list. For example, you can plot the

spirals r =
θ

2π
, r =

(
θ

2π

)2

, and the circle r = 1 all in the same picture with:

In[ ]: PolarPlot[{theta/(2Pi), (theta/(2Pi))^2, 1}, {theta, 0, 2Pi}]

Out[ ]:

Plotting Graphs of Implicit Equations When a curve is defined by an equation in two
variables, you can draw the curve with the ContourPlot,a command whose full functionality
we will discuss more carefully later. For now, we will use ContourPlot to draw just that
portion of the curve defined by an equation in, say, the two variables x and y, which lies
within a rectangle x ∈ [a, b] and y ∈ [c, d] with the syntax:

ContourPlot[ equation of x and y, {x,a,b}, {y,c,d}]

For example, to see the unit circle x2 + y2 = 1 which we know lies within the rectangle
[−1, 1]× [−1, 1], we use:

In[ ]: ContourPlot[x^2+y^2==1, {x,-1,1}, {y,-1,1}]

Out[ ]:

You can plot several curves defined by equations in a single command. The syntax is exactly
the same as it is for the Plot, ParametricPlot, and PolarPlot commands. For exam-
ple, we will plot the portions of the curves x2 + 3xy+ y3 = 25, x2 + 3xy+ y3 = 10, and
x2 + 3xy+ y3 = 0 which lie within the rectangle −10 ≤ x ≤ 10 and −7 ≤ y ≤ 4. We will
remove the frame, and draw the axes within the picture itself.

In[ ]: f[x , y ] := x^2 + 3 x y + y^3;

In[ ]: ContourPlot[{f[x,y]==25, f[x,y]==10, f[x,y]==0},
{x,-10,10},{y,-7,4}, Frame->False, Axes->True]

Out[ ]:

An interesting feature of the default output from ContourPlot is that if you position the
mouse over any of the three curves shown in the graphic above, you will see the equation
shown as a tool tip on the mouse.

Tick Marks Each of the commands Plot, ParametricPlot, PolarPlot, and ContourPlot
allows you to control the placement of tick marks on each of the axes using the Ticks option.
This is done as we do in the following example. Suppose we draw the graph of x+ sin(3x),

and we wish to have tick marks on the x−axis at
π

6
,
2π

6
,
3π

6
,
4π

6
,
5π

6
and on the y−axis at

1, 2, 3.

In[ ]: Plot[x+Sin[3x], {x,0,Pi}, Ticks->{Range[5]Pi/6, {1,2,3}}]

Out[ ]:

You can specify the labels to be shown at given tick marks.

In[ ]: Plot[x+Sin[3x],{x,0,Pi},Ticks->{Range[5]Pi/6,{{1,"A"},{2,"B"},{3,"C"}}}]

Out[ ]:

Labels The AxesLabel and PlotLabel options allow you to specify labels for one or both
of the axes, or for the output graphic as a whole, respectively.

In[ ]: Plot[Cos[x]+15, {x,0,4Pi}, AxesLabel->{"day","price"},
PlotLabel->"Daily Stock Price"]

Out[ ]:

If your picture is framed, then you must use the FrameLabel option in place of AxesLabel.

In[ ]: ContourPlot[x^2/16+y^2/9==1, {x,-4,4}, {y,-3,3}, Axes->True,

AspectRatio->Automatic, FrameLabel->{"x-axis","y-axis"},
PlotLabel->"Ellipse"]

Out[ ]:

9.3 Making Graphs in Space

The easiest way to sketch a surface in three dimensions (3D) is to use the Plot3D command.
You input an expression that gives the height of a surface above the xy−plane, in terms of the

independent variables x and y. You must also specify intervals x ∈ [x0, x1] and y ∈ [y0, y1].
The Plot3D command then has the form:

Plot3D[ expression of x and y, {x,x0,x1}, {y,y0,y1}]

For example, The surface whose height is z = 4− x2 − y2 above the xy−plane, over the
rectangle [−2, 2]× [−2, 2], is seen with:

In[ ]: Plot3D[4-x^2-y^2, {x,-2,2}, {y,-2,2}]

Out[ ]:

The graph of the function f(x, y) = x2 − y2 looks like a saddle if we take, for example,
x, y ∈ [−3, 3] as shown below.

In[ ]: Plot3D[x^2-y^2, {x,-3,3}, {y,-3,3}]

Out[ ]:

There are a number of options that you can use to add more character to the graph of a
function z = f(x, y) with Plot3D. For example,

1. AxesLabel : Provides names to label each of the three axes.

2. BoxRatios : Scales the graphic so that it appears within a certain lengths of a box.

3. PlotRange : Specify the range of one or more of the variables x, y, and z.

4. Mesh→False : Omit the mesh on the graph.

5. PlotStyle→Green : Color the graph green.

For example see how the sombrero looks like using some of Plot3D options. Note that the

sombrero is given by the function f(x, y) =
sin
(√

x2 + y2
)

√
x2 + y2

.

In[ ]: Plot3D[Sin[Sqrt[x^2+y^2]]/Sqrt[x^2+y^2], {x,-7,7},
{y,-7,7}, PlotPoints->50, PlotRange->All, AxesLabel->{"x","y","z"},
Mesh->False, BoxRatios->{1,1,1}]

Out[ ]:

Mathematica includes very useful Notebook feature for working with 3D graphics. As you
move over a 3D graphic with the mouse, you will see that the cursor changes. If you now
click and drag with the mouse, you will be able to rotate the graphic in real time. This lets
you view the surface from many different angles. Every three-dimensional picture drawn in
Mathematica will be shown from a specified ViewPoint. Specifying a ViewPoint is the same
as describing how your eye is located with respect to the graphic being drawn. Let us show
you how it is done. You input the ViewPoint option in the form ViewPoint->{a,b,c},
where the values a, b, and c describe the position of your eye relative to the object that you
are viewing. Positive values of a, b, and c place you ”in front of”, ”to the right of”, and
”above” the object, with respect to the direction of the positive x-, y- and z-coordinate axes,
respectively. Negative values place you ”behind”, ”to the left of”, and ”below” the object.
A value of zero for any of these three places you ”at the center”. Let us think of the graph

of f(x, y) =
(
x2 + y2

)
e−

x2

400
− y2

100 as being twin mountains. Let us take a ”helicopter ride”
around these mountains using three ViewPoints: {3,1,1}, {3,2,-1}, and {0,100,0},
respectively.

In[ ]: f[x ,y ] := (x^2+y^2) Exp[-x^2/400 - y^2/100];

In[ ]: vp1=Plot3D[f[x, y],{x,-40,40},{y,-40,40},ViewPoint->{3,1,1}];
In[ ]: vp2=Plot3D[f[x, y],{x,-40,40},{y,-40,40},ViewPoint->{3,2,-1}];
In[ ]: vp3=Plot3D[f[x, y],{x,-40,40},{y,-40,40},ViewPoint->{0,100,0}];
In[ ]: GraphicsRow[{vp1, vp2, vp3}, ImageSize -> Full]

Out[ ]:

9.4 Surfaces in Cylindrical and Spherical Coordinates

Surfaces are sometimes more easily described in terms of the cylindrical or spherical co-
ordinate systems. You can draw such a surface easily with the Mathematica commands
RevolutionPlot3D and SphericalPlot3D commands, respectively.
Points in the cylindrical coordinate system are described by quantities r, θ, and z, where

r is the horizontal radial distance of the point from the z−axis;

θ is the horizontal angle measured from the x−axis; and

z is the z−coordinate in standard rectangular coordinates.

To draw the surface z = f(r, θ) for r ∈ [r0, r1] and θ ∈ [θ0, θ1], you enter:

RevolutionPlot3D[f[r,θ],{r,r1,r2},{θ,θ1,θ2}]

For example, to see the surface z = r2 cos(8θ), for r ∈ [0.5, 1] and θ ∈ [0, 2π]:

In[ ]: RevolutionPlot3D[r^2 Cos[8 theta], {r,0.5,1}, {theta,0,2Pi}]

Out[ ]:

In Mathematica, points in the spherical coordinate system are described by quantities ρ, ϕ,
and θ, where

ρ is the radial distance in space of the point from the origin;

θ is the vertical angle measured from the positive z−axis; and

ϕ is the horizontal angle measured from the x−axis.

Note that Mathematica uses ϕ and θ for the horizontal and vertical angles, respectively,
in spherical coordinates. This is exactly the opposite of the notation used in almost every
Calculus book.

To draw the surface ρ = f(θ, ϕ), where θ ∈ [θ0, θ1] and ϕ ∈ [ϕ0, ϕ1], you will enter:

SphericalPlot3D[f[θ,ϕ],{θ,θ1,θ2},{ϕ,ϕ1,ϕ2}]]

For example, to see the surface defined by ρ = cos(θ) cos(4θ), for θ ∈
[
0, π

4

]
, and ϕ ∈ [0, 2π],

use:

In[ ]: SphericalPlot3D[Cos[theta]Cos[4theta], {theta,0,Pi/4},{phi,0,2Pi}]

Out[ ]:

When you use the SphericalPlot3D command, you must enter the interval for the vertical
angle theta first, then the interval for the horizontal angle phi. If you do not use this order,
the picture will be incorrect.

Plotting Multiple Surfaces The Plot3D command allows you to draw more than one
surface above a rectangle x ∈ [a, b] and y ∈ [c, d] in a single graphic. You use it in the form

Plot3D[{surface1, surface2, ...},{x,a,b},{y,c,d}]

and you still can attach options just as before. For example, the intersection of the paraboloid
z = x2 + y2 with the plane z = 2x+ 5y is a circle. You can see this quite easily with

In[ ]: Plot3D[{x^2+y^2, 2x+5y}, {x,-6,6}, {y,-6,6}]

Out[ ]:

SphericalPlot3D similarly allows you to plot more than one surface, using the same syntax
extension as the Plot3D command uses. For example, we can see a very nice ”cut-away”
image of the 3 spheres r = cos(θ), r = 2 cos(θ), and r = 3 cos(θ) by restricting the range

of ϕ, to stay between
π

4
and

7π

4
and looking at them from in front of the x−axis:

In[ ]: SphericalPlot3D[{Cos[theta], 2Cos[theta], 3Cos[theta]},
{theta,0,Pi/2}, {phi,Pi/4,7Pi/4}, ViewPoint->{2, 0, 0}]

Out[ ]:

Finally, although RevolutionPlot3D also allows multiple equations in the same graphic,
the syntax is different. You must use one extra layer of curly braces { and }. For example,
in cylindrical coordinates, the equation z = r, for 1 ≤ r ≤ 2 and 0 ≤ θ ≤ 2π, describes a
portion of a cone, also the equations z = r± 1 are just vertical translations of the cone. Put
the three together and you have got three bowls stacked up.

In[ ]: RevolutionPlot3D[{{r}, {r-1}, {r+1}}, {r,0,2}, {theta,0,2Pi}]

Out[ ]:

9.5 Changing Coordinate Systems

Changing coordinate systems can involve two very different operations. One is recomputing
coordinate values that correspond to the same point. The other is re-expressing a field in
terms of new variables. The Wolfram Language provides functions to perform both these
operations.

Converting Points Two coordinate systems are related by a mapping that takes coordinate
values in the old system and returns coordinate values in the new system. The function
CoordinateTransformData returns information about mappings between the coordinate
systems. For example, the following converts the point (r, θ) in polar coordinates to the
corresponding (x, y) Cartesian coordinates.

In[ ]: CoordinateTransformData["Polar"->"Cartesian","Mapping",{r,θ}]
Out[ ]: {r cos(θ), r sin(θ)}

Conversely, we write

In[ ]: CoordinateTransformData["Cartesian"->"Polar","Mapping",{x,y}]

Out[ ]:
{√

x2 + y2, tan−1
(y
x

)}
To covert a point (ρ, θ, ϕ) in spherical coordinates to the corresponding (x, y, z) Cartesian
coordinates, and vice versa, we use

In[ ]: CoordinateTransformData["Spherical"->"Cartesian","Mapping", {ρ, θ, ϕ}]
Out[ ]: {ρ sin(θ) cos(ϕ), ρ sin(θ) sin(ϕ), ρ cos(θ)}

In[ ]: CoordinateTransformData["Cartesian"->"Spherical","Mapping",{x,y,z}]

Out[ ]:

{√
x2 + y2 + z2, tan−1

(√
x2 + y2

z

)
, tan−1

(y
x

)}

Also, we convert a point (r, θ, z) in cylindrical coordinates to the corresponding (x, y, z)
Cartesian coordinates, and vice versa by writing

In[ ]: CoordinateTransformData["Cylindrical"->"Cartesian","Mapping", {r,θ,z}]
Out[ ]: {r cos(θ), r sin(θ), z}
In[ ]: CoordinateTransformData["Cartesian"->"Cylindrical","Mapping", {x,y,z}]

Out[ ]:
{√

x2 + y2, tan−1
(y
x

)
, z
}

Example 9.1. Convert the point (1, 1, 0) in Cartesian coordinates to the corresponding
values in spherical coordinates.

In[ ]: CoordinateTransformData["Cartesian"->"Spherical","Mapping",{1,1,0}]

Out[ ]:
{√

2,
π

2
,
π

4

}

Example 9.2. Convert the point
(
1,

π

3
,
π

4

)
in spherical coordinates to the corresponding

cylindrical coordinates.

In[ ]: CoordinateTransform["Spherical"->"Cylindrical",{1,Pi/3,Pi/4}]

Out[ ]:

{√
3

2
,
π

4
,
1

2

}

Example 9.3. Convert the points (1, 1), (1,−1), and (0,−2) from Cartesian coordinates to
the corresponding polar coordinates.

In[ ]: CoordinateTransform["Cartesian"->"Polar",{{1,1},{1,-1},{0,-2}}]

Out[ ]:
{{√

2,
π

4

}
,
{√

2,−π

4

}
,
{
2,−π

2

}}

Transforming Fields When transforming fields between two coordinate systems, a field
given in terms of variables in the old system is re-expressed in terms of variables in the new
system. In addition to the mapping between the systems, several additional steps are needed:
solving for the old variables in terms of the new, the substituting in these expressions. All of
these steps are performed by the command TransformedField. For example, this converts
the scalar field x2 + y2 + z2 from Cartesian (x, y, z) to cylindrical (r, θ, ξ) coordinates.

In[ ]: TransformedField["Cartesian"->"Cylindrical", x^2+y^2+z^2, {x,y,z}->
{r,θ,ξ}] // Simplify

Out[ ]: ξ2 + r2

The following converts the hyperbola x2 − y2 from Cartesian (x, y) to polar (r, θ) coordi-
nates.

In[ ]: TransformedField["Cartesian"->"Polar",x^2-y^2, {x,y}->{r,θ}]//Simplify
Out[ ]: r2 cos(2θ)

Example 9.4. In some cases, a surface given in rectangular coordinates will look better if
you draw it using cylindrical or spherical coordinates. For example, the surface defined by

the equation z =
x2 − y2

(x2 + y2)2
can be plotted with:

In[ ]: Plot3D[(x^2-y^2)/(x^2+y^2)^2,{x,-3,3},{y,-3,3}]

Out[ ]:

The picture is choppy especially near the origin. However, if we use cylindrical coordinates,
the surface becomes

In[ ]: f=TransformedField["Cartesian"->"Cylindrical",

(x^2-y^2)/(x^2+y^2)^2, {x,y,z}->{r,θ,ξ}] // Simplify

Out[ ]:
cos(2θ)

r2

Now, we will do the plot.

In[ ]: RevolutionPlot3D[f,{r,0.5,3},{θ,0,2Pi}]

Out[ ]:

9.6 Level Curves and Level Surfaces

In Mathematica, the level curves (contours) of a function f(x,y) are plotted with the
ContourPlot command. To see the level curves inside the rectangle x ∈ [a, b] and y ∈ [c, d],
you use the command with this syntax:

ContourPlot[f[x,y],{x,a,b},{y,c,d}]

For example, here are some level curves of f(x, y) = xye−x2−y2 near the origin:

In[ ]: ContourPlot[x y Exp[-x^2-y^2], {x,-2,2}, {y,-2,2}]

Out[ ]:

By default, Mathematica automatically sketches an appropriate number of level curves and
colors the areas between these curves. Lighter shades represent higher levels, while darker
shades represent lower levels. By moving the mouse over the graphic above, Mathematica
will show you (using a tooltip at the cursor) exactly what contour levels have been drawn.

Level Surfaces in Space If f(x, y, z) is a function of three variables defined over a region
x ∈ [x0, x1], y ∈ [y0, y1], and z ∈ [z0, z1] then the level surface of f at level c can be seen
with the ContourPlot3D command.

ContourPlot3D[function, {x,x0,x1}, {y,y0,y1}, {z,z0,z1}, Contours->{c}]

You can specify more than one level surface to be shown in the same graphic by writing
Contours→{the levels}, where the levels are separated by commas. For example, We
can see the level surfaces of f(x, y, z) = x3 − y2 + z2 at the levels of 1 and 10 with the
following.

In[ ]: ContourPlot3D[x^3-y^2+z^2,{x,-2,3},{y,-2,2},{z,-2,3}, Contours->{1,10}]

Out[ ]:

9.7 Parametric Curves and Surfaces in Space

You use the ParametricPlot3D command to draw a space curve. To see the curve given
parametrically as

(
x(t), y(t), z(t)

)
, for t ∈ [a, b], type:

92



Feras Awad @2025 9 GRAPHICS IN MATHEMATICA

ParametricPlot3D[{x(t),y(t),z(t)},{t,a,b}]

This format is similar to the ParametricPlot command used for plane curves. Here, how-
ever, the curve is defined with three parametric functions rather than two. For example, the
helix, given parametrically by (t, 3 cos t, 3 sin t), for t ∈ [0, 8π], is drawn with:

In[ ]: ParametricPlot3D[{t,3Cos[t],3Sin[t]}, {t,0,8Pi}]

Out[ ]:

ParametricPlot3D can also be used to draw a surface in space. If a surface is defined
parametrically by

(
x(u, v), y(u, v), z(u, v)

)
for u ∈ [a, b] and v ∈ [c, d], you enter:

ParametricPlot3D[{x(u,v),y(u,v),z(u,v)},{u,a,b},{v,c,d}]

For example, to see a portion of the one-sheeted hyperboloid given parametrically by
(
cos u cosh v,

sin u cosh v, sinh v
)
, for u ∈ [0, 2π] and v ∈ [−2, 2], write:

In[ ]: ParametricPlot3D[{Cos[u]Cosh[v],Sin[u]Cosh[v],Sinh[v]}, {u,0,2Pi},
{v,-2,2}]

Out[ ]:

9.8 Visualizing Data

In Mathematica, curly braces are used to represent lists, regardless of the type of elements
in the list. Lists can be created several different ways as we have seen before in chapter 4.

It is common to store lists in variables; this allows the lists to be easily referenced in subse-
quent calculations. For example, by assigning a list to the symbol data, this variable can be
used in other calculations or commands where the list is needed.

In[ ]: data = {8, 2, 5, 1, 4, 7, 6}
Out[ ]: {8, 2, 5, 1, 4, 7, 6}

Just as the Wolfram Language has many commands available to visualize all types of math-
ematical functions and surfaces, so too does it have many commands available to visualize
lists and datasets. One of the most common commands to visualize data is ListPlot, which
displays the data as individual points.

In[ ]: ListPlot[data]

Out[ ]:

When values jump around, it is usually harder to understand if you do not join them up.
ListLinePlot plots a list, joining up values.

In[ ]: ListLinePlot[data]

Out[ ]:

If you just want to know which numbers appear, you can plot them on a number line.

In[ ]: NumberLinePlot[data]

Out[ ]:

Making a bar chart can be useful too:

In[ ]: BarChart[data]

Out[ ]:

So long as the list is not too long, a pie chart can be useful:

In[ ]: PieChart[data]

Out[ ]:

Multiple Datasets While the examples thus far have concentrated on visualizing a single
dataset, the visualization commands can also be used to visualize multiple datasets. Multiple
datasets can be constructed by placing several different lists into a single ”parent” list that
encompasses them all. This larger list can be passed to a command like ListPlot to visualize
each of the sublists as a separate dataset. The following example shows the use of ListPlot
to plot two datasets, which are automatically given different colors to easily tell them apart.

In[ ]: ListPlot[{{3, 5, 7, 9}, {1, 4, 9, 16, 25}}]

Out[ ]:

In[ ]: ListLinePlot[{{3, 5, 7, 9}, {1, 4, 9, 16, 25}}]

Out[ ]:

The following example depicts a use of the Show command to combine a data visualization
from ListPlot and a function visualization from Plot.

In[ ]: Show[Plot[x^2, {x,0,5}], ListPlot[{1,4,9,16,25},PlotStyle->
{Red,PointSize[Large]}]]

Out[ ]:

95



Feras Awad @2025 9 GRAPHICS IN MATHEMATICA

Note that we add an option to ListPlot called PlotStyle, which allows us to color the
data points in red and change their size to large, in order to make them stand out.

Many measurements are commonly represented as two-dimensional datasets. The data vi-
sualization commands in the Wolfram Language are designed to work with both one- and
multidimensional data. When a command like ListPlot is given a one-dimensional list,
it is assumed that the list contains y values that correspond to x values 1, 2, and so on.
ListPlot also accepts a list of (x,y) pairs instead of single height values for y coordinates.

In[ ]: ListPlot[Table[{2i,Prime[i]}, {i,8}]]

Out[ ]:

Visualizing lists or datasets in three dimensions is just as easy as visualizing them in one
or two dimensions. Many of the plotting commands shown thus far have 3D equivalents,
like ListPlot and ListPlot3D. The ListPlot3D command visualizes a 3D surface based
on the values from that dataset, and this 3D surface has the same interactivity (rotation,
panning, zooming) as other objects.

In[ ]: ListPlot3D@Table[Cos[i] Cos[j], {i,2,8},{j,-3,1}]

Out[ ]:

To visualize only the discrete data points and not a connecting mesh between the points,
ListPointPlot3D can be used.

In[ ]: ListPointPlot3D@Table[Cos[i] Cos[j], {i,2,8},{j,-3,1}]

Out[ ]:

Visualize a Matrix Vectors and matrices have special visualization commands. ArrayPlot
draws a representation of an array, coloring squares that represent larger values with darker
colors. The following is the ArrayPlot of the 8× 8 matrix of the first 64 consecutive prime
numbers.

In[ ]: ArrayPlot[Partition[Prime[Range[64]],8]]

Out[ ]:

MatrixPlot follows a similar logic, where the position of a value in a matrix determines the
coloration of that position in the plot, with negative values shown in cool tones, like blue,
and positive values shown in warm tones, like orange. The higher the magnitude of a value,
the more intense its corresponding color is for that position.

In[ ]: MatrixPlot[{{-10,-5,-1}, {2,4,6}, {20,30,40}}]

Out[ ]:

Shape Matrices Mathematica provides certain shapes of matrices as follows.

1. DiskMatrix[r] gives a matrix whose elements are 1 in a disk-shaped region of radius
r, and are otherwise 0.

In[ ]: ArrayPlot[DiskMatrix[7], Mesh -> All]

Out[ ]:

2. BoxMatrix[r] gives a (2r+ 1)× (2r+ 1) matrix of 1’s.

In[ ]: ArrayPlot[BoxMatrix[7], Mesh -> All]

Out[ ]:

3. CrossMatrix[r] gives a matrix whose elements are 1 in a centered cross-shaped
region that extends r positions along each index direction, and are 0 otherwise.

In[ ]: ArrayPlot[CrossMatrix[7], Mesh -> All]

Out[ ]:

4. DiamondMatrix[r] gives a matrix whose elements are 1 in a diamond-shaped region
that extends r index positions to each side, and are 0 otherwise.

In[ ]: ArrayPlot[DiamondMatrix[7], Mesh -> All]

Out[ ]:

9.9 Advanced Graphics

In addition to using commands such as Plot and ParametricPlot to create 2D pictures,
we can ask Mathematica to directly draw simple objects such as points, lines, circles, and
rectangles. These objects are called graphics primitives, and they are the lowest-level com-
mands used to produce every 2D Mathematica picture. To work with graphics primitives,
you have to first define them with the Graphics command:

Graphics[ graphics primitive ]

or, to group several graphic primitives together, use

Graphics[ {graphics primitive1, graphics primitive2, ...} ]

3D graphics primitives are also available in Mathematica. You use the Graphics3D command
both to define and show them. Some of the graphics primitives we have used often are the
following.

1. Point[p] is a graphics and geometry primitive that represents a point at p.

In[ ]: Graphics[Point[{-2,1}]]

Out[ ]:

In[ ]: Graphics[Point[Table[{t,Sin[t]}, {t,0,2Pi, 2Pi/10}]]]

Out[ ]:

In[ ]: pts=Table[{t,Sin[t],Cos[t]},{t,0,2Pi,2Pi/10}];
In[ ]: Graphics3D[Point[pts]]

Out[ ]:

2. Line[{p1,p2,...}] represents the line segments joining a sequence for points pi.

In[ ]: Graphics[{Thick, Line[{{1,0}, {2,1}, {3,0}, {4,1}}]}]

Out[ ]:

In[ ]: line=Line[{{1,1,-1},{2,2,1},{3,3,-1},{4,4,1}}];
In[ ]: Graphics3D[{Thick,Dashed,line}]

Out[ ]:

3. InfiniteLine[{p1,p2}] represents the infinite straight line passing through the
points p1 and p2.

In[ ]: Graphics[InfiniteLine[{0,0},{1,1}], Frame->True]

Out[ ]:

In[ ]: Graphics3D[InfiniteLine[{0,0,0},{1,2,3}]]

Out[ ]:

In[ ]: lines=Table[InfiniteLine[{0,y},{1,1}],{y,5}];
In[ ]: Graphics[lines,PlotRange->{{-6,6},{0,6}}]

Out[ ]:

Example 9.5. Find the point of intersection for the two lines ℓ1 that passes
through the point (0, 0) in the direction of the vector (1, 1), and the line ℓ2 that
passes through the points (0, 1), (1, 0).

In[ ]: line1 = InfiniteLine[{0,0},{1,1}];
In[ ]: line2 = InfiniteLine[0, 1, 1, 0];

In[ ]: sol=Solve[Element[{x,y},line1]&& Element[{x,y},line2],{x,y}

Out[ ]:

{{
x → 1

2
, y → 1

2

}}
In[ ]: Graphics[{{Thick,line1,line2},{PointSize[0.06],Red,

Point[{x,y}/.sol]}},Axes->True,ImageSize->Small]

Out[ ]:

4. Arrow[{pt1,pt2}] is a graphics primitive that represents an arrow from pt1 to pt2.

In[ ]: Graphics[Arrow[{{1,0},{2,1},{3,0},{4,1}}]]

Out[ ]:

In[ ]: Plot[Sin[x],{x,0,2Pi},Epilog->{Arrow[{{3Pi/2,1/2},{Pi,0}}],
Text["Zero",{3Pi/2,1/2},{-1,-1}]}]

Out[ ]:

5. Triangle[{p1,p2,p3}] represents a filled triangle with corner points p1, p2, and p3.

In[ ]: Graphics[Triangle[]]

Out[ ]:

In[ ]: Graphics3D[Triangle[{{0,0,0},{1,0,0},{0,1,1}}]]

Out[ ]:

Example 9.6. To define an equilateral triangle by side length s, we use the com-
mand SSSTriangle[a,b,c] which returns a filled triangle with sides of lengths
a, b, and c.

In[ ]: EquilateralTriangle[s ] := SSSTriangle[s,s,s]

In[ ]: EquilateralTriangle[2]

Out[ ]: Triangle[{{0, 0}, {2, 0}, {1,
√
3}}]

In[ ]: Graphics@EquilateralTriangle[2]

Out[ ]:

Example 9.7. To define an isosceles triangle by side length s that adjacent to
both equals angles α, we may use either the command IsoscelesTriangle[a,s,b]
which returns a filled triangle with angles a and b and side length c, and c is
adjacent to both angles, or we use the command SASTriangle[s1,a,s2] which
returns a filled triangle with sides of length s1 and s1 and angle a between them.

In[ ]: IsoscelesTriangle[a ,s ,b ] := ASATriangle[a,s,b]

In[ ]: IsoscelesTriangle[Pi/6,3,Pi/6]

Out[ ]: Triangle[

{
{0, 0}, {3, 0},

{
3

2
,

√
3

2

}}
]

In[ ]: Graphics@IsoscelesTriangle[Pi/6,3,Pi/6]

Out[ ]:

In[ ]: IsoscelesTriangle[s1 ,a ,s2 ] := SASTriangle[s1,a,s2]

In[ ]: Graphics@IsoscelesTriangle[2, Pi/6, 2]

Out[ ]:

6. Rectangle[{xmin,ymin},{xmax,ymax}] represents an axis-aligned filled rectangle
from {xmin, ymin} to {xmax, ymax}.

In[ ]: Graphics[Rectangle[]] (* this command defines a unit square *)

Out[ ]:

In[ ]: Graphics[Rectangle[{2,1},{4,5}],Axes->True,AxesOrigin->{0,0}]

Out[ ]:

7. RegularPolygon[n] gives the regular polygon with n vertices equally spaced around
the unit circle. RegularPolygon[n,r] gives the regular polygon of radius r.

In[ ]: Table[Graphics[RegularPolygon[k]],{k,3,8}]

Out[ ]:

8. Circle[{x,y},r] represents a circle of radius r centered at {x,y}. The command
Circle[{x,y},{rx,ry}] gives an axis-aligned ellipse with semi-axes lengths rx and
ry. While Circle[{x,y},...,{θ1,θ2}] gives a circular or ellipse arc from angle θ1
to θ2.

In[ ]: Graphics[Circle[]] (* Unit Circle *)

Out[ ]:

In[ ]: Graphics[Circle[{0,0},1,{Pi/6,3Pi/4}]] (* Circular Arc *)

Out[ ]:

In[ ]: Graphics[Circle[{0,0},{3,4}]] (* An Ellipse *)

Out[ ]:

In[ ]: Graphics[Table[Circle[{i,j},1/2],{i,7},{j,5}]]

Out[ ]:

9. Disk[{x,y},r] represents a disk of radius r centered at {x,y}.

In[ ]: Graphics[Disk[]] (* Unit Disk *)

Out[ ]:

In[ ]: Graphics[{Orange,Disk[{0,0},1,{Pi/4,3Pi/4}]}] (* Disk Sector *)

Out[ ]:

In[ ]: Graphics[{Gray,Disk[{0,0},{3,4}]}] (* An Elliptical Disk *)

Out[ ]:

10. InfinitePlane[{p1,p2,p3}] represents the plane passing through the points p1,
p2, and p3. InfinitePlane[p,{v1,v2}] represents the plane passing through the
point p in the directions v1 and v2.

To represent a plane through the points (1,0,0), (1,1,1), and (0,0,1)

In[ ]: Graphics3D[InfinitePlane[{{1,0,0},{1,1,1},{0,0,1}}]]

Out[ ]:

To define the plane passing through the point (1,0,0) in the directions (0,1,1)
and (1,0,1)

In[ ]: Graphics3D[InfinitePlane[{1,0,0},{{0,1,1},{-1,0,1}}]]}]

Out[ ]:

11. Ball[p,r] represents a ball of radius r centered at the point p.

A unit ball at the origin:

In[ ]: Graphics3D[Ball[{0,0,0}]]

Out[ ]:

12. Cylinder[{{x1,y1,z1},{x2,y2,z2}},r] represents a cylinder of radius r around
the line from (x1,y1,z1) to (x2,y2,z2).

Cylinder from the origin to {1,1,1} with radius
1

2
.

In[ ]: Graphics3D[Cylinder[{{0,0,0},{1,1,1}},1/2]]

Out[ ]:

13. Cone[{{x1,y1,z1},{x2,y2,z2}},r] represents a cone with a base of radius r cen-
tered at (x1,y1,z1) and a tip at (x2,y2,z2).

Cone from the origin to {1,1,1} with radius
1

2
at its base

108



Feras Awad @2025 9 GRAPHICS IN MATHEMATICA

In[ ]: Graphics3D[Cone[{{0,0,0},{1,1,1}},1/2]]

Out[ ]:

14. Cuboid[pmin,pmax] represents an axis-aligned filled cuboid with lower corner pmin
and upper corner pmax.

In[ ]: Graphics3D[Cuboid[{-2,0,1},{1,3,3}],Axes->True,AxesOrigin->{0,0,0}]

Out[ ]:

9.10 Region Integrals Measures

The Wolfram Language supports a broad range of standard properties and measures for
geometric regions, including integral measures such as length, area, volume, and centroid.

Arc Length The command ArcLength[reg] gives the length of the one dimensional

region reg. For example, the arc length of the function f(x) =
x3

6
− 1

2x
on the interval

[1, 2] is

In[ ]: ArcLength[x^3/6+1/(2x), {x,1,2}]

Out[ ]:
17

12

The length can be computed using the polar representation of r(θ) = 2− 2 cos θ on [0, 2π]
as follows.

In[ ]: ArcLength[{2-2Cos[theta],theta},{theta,0,2Pi},"Polar"]
Out[ ]: 16

Length of one revolution of the helix given parametrically is

In[ ]: ArcLength[{Cos[t],Sin[t],t},{t,0,2Pi}]
Out[ ]: 2

√
2π

The length of the line connecting the points (0, 0), (1, 1), and (3,−1) is

In[ ]: ArcLength[Line[{{0,0},{1,1},{3,-1}}]]
Out[ ]: 3

√
2

The length of a circle with radius r is

In[ ]: ArcLength[Circle[{x,y},r]]
Out[ ]: 2πr

The length of an ellipse is

In[ ]: N@ArcLength[Circle[{0,0},{2,3}]]
Out[ ]: 15.8654

The arc length of a circle intersected with a triangle:

In[ ]: R1 = Circle[];

In[ ]: R2 = Triangle[{{-3/2,0},{3/2,0},{0,11/10}}];
In[ ]: R3 = RegionIntersection[R1,R2];

Out[ ]:

In[ ]: ArcLength[R3]

Out[ ]: 2

(
sin−1

(
3

692

(
121− 5

√
295
))

+ cos−1

(
3

692

(
5
√
295+ 121

)))
The perimeter of the triangle whose two of its angles are

π

6
and

π

3
, and the length of the

side adjacent to both angles is of length 2, is

In[ ]: ArcLength[RegionBoundary[ASATriangle[Pi/6,2,Pi/3]]]]

Out[ ]:
√
3+ 3

where the command RegionBoundary represents the boundary of the region reg.

Area The command Area[reg] gives the area of the two dimensional region reg. For
example, the area of the disk with radius r is

In[ ]: Area[Disk[{x,y}, r]]

Out[ ]: πr2

The surface area of a sphere (ball) with radius r is

In[ ]: Area[RegionBoundary[Ball[{x,y,z},r]]]
In[ ]: Area[Sphere[{x,y,z},r]]
Out[ ]: 4πr2

The area of an annulus with inner radius 1 and outer radius 2 equals

In[ ]: Area[{r Sin[theta],r Cos[theta]},{r,1,2},{theta,0,2Pi}]
Out[ ]: 3π

The area of the triangle whose vertices at the points
(
−1

2
,−1

)
, (1, 2), and (3, 0) is

In[ ]: Area[Triangle[{{-1/2,-1},{1,2},{3,0}}]]

Out[ ]:
9

2

The area of a triangle with sides of length a, b, and c is

In[ ]: Area[SSSTriangle[a,b,c]]

Out[ ]:
1

4

√
|(a+ b− c)(a− b+ c)(−a+ b+ c)(a+ b+ c)|

The area of an ellipse centered at
(
cx, cy

)
is

In[ ]: Area[Disk[{cx,cy},{a,b}]]
Out[ ]: abπ

Volume The command Volume[reg] gives the volume of the three dimensional region
reg. For example, the volume of a ball in 3D of radius r is

In[ ]: Volume[Ball[{a,b,c},r]]

Out[ ]:
4πr3

3

The volume and the surface area of a cone of height h and radius r of its circular base are

In[ ]: PowerExpand@Volume[Cone[{{0,0,0},{0,0,h}},r]]

Out[ ]:
1

3
πhr2

In[ ]: Area[RegionBoundary[Cone[{{0,0,0},{0,0,h}},r]]]

Out[ ]: πr
(√

h2 + r2 + r
)

Centroid The centroid or geometric center of a plane figure is the arithmetic mean (”av-
erage”) position of all the points in the shape. RegionCentroid[reg] gives the centroid
of the region reg. For example, the centroid of the disk with center (a,b) and radius r is

In[ ]: RegionCentroid[Disk[{a,b},r]]
Out[ ]: {a, b}

The centroid of the rectangle is

In[ ]: RegionCentroid[Rectangle[{a,b},{c,d}]]

Out[ ]:

{
a+ c

2
,
b+ d

2

}
9.11 Graphs and Networks

A graph is a way of showing connections between things - say, how web pages are linked, or
how people form a social network. Let us start with a very simple graph, in which 1 connects
to 2, 2 to 3, and 3 to 4. Each of the connections is represented by → (typed as ->).

In[ ]: Graph[{1->2,2->3,3->4}]
Out[ ]:

Automatically label all the ”vertices”:

In[ ]: Graph[{1->2,2->3,3->4},VertexLabels->All]

Out[ ]:

Let us add one more connection: to connect 4 to 1. Now we have a loop.

In[ ]: Graph[{1->2,2->3,3->4,4->1},VertexLabels->All]

Out[ ]:

Add two more connections, including one connecting 2 right back to 2:

In[ ]: Graph[{1->2,2->3,3->4,4->1,3->1,2->2},VertexLabels->All]

Out[ ]:

As we add connections, the Wolfram Language chooses to place the vertices or nodes of the
graph differently. All that really matters for the meaning, however, is how the vertices are
connected. And if you do not specify otherwise, the Wolfram Language will try to lay the
graph out so it is as untangled and easy to understand as possible.

You can do computations on the graph, say finding the shortest path that gets from 4 to 2,
always following the arrows.

In[ ]: FindShortestPath[Graph[{1->2,2->3,3->4,4->1,3->1,2->2}],4,2]
Out[ ]: {4, 1, 2}

Now let us make another graph. This time let us have 3 nodes, and let us have a connection
between every one of them. Start by making an array of all possible connections between 3
objects:

In[ ]: Table[i->j,{i,3},{j,3}]
Out[ ]: {{1 → 1, 1 → 2, 1 → 3}, {2 → 1, 2 → 2, 2 → 3}, {3 → 1, 3 → 2, 3 → 3}}

The result here is a list of lists. But what Graph needs is just a single list of connections.
We can get that by using Flatten to ”flatten” out the sublists.

In[ ]: Flatten[Table[i->j,{i,3},{j,3}]]
Out[ ]: {1 → 1, 1 → 2, 1 → 3, 2 → 1, 2 → 2, 2 → 3, 3 → 1, 3 → 2, 3 → 3}

Now, show the graph of these connections:

In[ ]: Graph[Flatten[Table[i->j,{i,3},{j,3}]],VertexLabels->All]

Out[ ]:

The following generates the completely connected graph with 6 nodes:

In[ ]: Graph[Flatten[Table[i->j,{i,6},{j,6}]]]

Out[ ]:

Sometimes the ”direction” of a connection does not matter, so we can drop the arrows.

In[ ]: UndirectedGraph[Flatten[Table[i->j,{i,6},{j,6}]]]

Out[ ]:

9.12 Exercises

1. The standard normal curve used in probability and statistics is defined by the function

f(x) =
1√
2π

e−x2/2. Sketch the graph of f for x ∈ [−3, 3].

2. The graphs of a function and its inverse are symmetric with respect to the line y = x.
Plot the functions f(x) = x2, x ∈ [0, 2], and its inverse f−1(x) =

√
x, x ∈ [0, 4], and

the line y = x and observe the symmetry.

3. Sketch the graphs of y = x2, y = −x2, and y = x2 sin(10x), x ∈ [−2π, 2π], on a single
set of axes enclosed by a frame.

4. Sketch the parabola y = x2 − 9 and a circle of radius 3 centered at the origin.

5. The curve traced by a point on a circle as the circle rolls along a straight line is called
a cycloid and has parametric equations x = r(θ − sin θ) and x = r(1− cos θ) where
r represents the radius of the circle. Plot the cycloid formed as a circle of radius 1
makes four complete revolutions.

6. The polar graph r = θ is called the Spiral of Archimedes. Sketch the graph for θ ∈
[0, 10π].

7. Sketch the graph defined by the equation y2 = x3(2− x) where x ∈ [0, 2] and y ∈ [−2, 2].

8. Plot the first 50 prime numbers.

9. Make a number line plot of the first 20 elements of the sequence given by the rule
1

n
.

10. Plot the functions y = x2 and y = 8− x2 and color the region enclosed between them
using the plot option Filling.

11. Plot the graph of the function e−x2−y2 above the rectangle [−2, 2]× [−2, 2].

12. Graph the paraboloid z = x2 + y2 with the plane z+ y = 12 above the rectangle
[−5, 5]× [−5, 5]. Do not draw axes or a surrounding box.

13. Sketch the space curve defined by the parametric equations

x(t) = (4+ sin 20t) cos t

y(t) = (4+ sin 20t) sin t

z(t) = cos 20t

where t ∈ [0, 2π]. This curve is called a toroidal spiral since it lies on the surface of a
torus.

14. Draw the ”ice cream cone” formed by the cone z = 3
√
x2 + y2 and the upper half of

the sphere x2 + y2 + (z− 9)2 = 9. Use cylindrical coordinates.

15. Sketch the graph of the surface ρ = 1+ sin 4θ sinϕ given in spherical coordinates
where θ ∈ [0, 2π] and ϕ ∈ [0, π].

16. Draw a contour plot of f(x, y) = sin x+ sin y on the square x, y ∈ [−4π, 4π].

17. Let f(x, y, z) = 5x2 + 2y2 + z2. Draw the level surfaces f(x, y, z) = k for k=1,4,9,16,
and 25. Sketch the surfaces only for y ≥ 0 so that all the surfaces will be visible.

18. A triangle has angles
π

6
and

π

4
, and a side of length 1 is adjacent to only one of

the angles. Find the perimeter of this triangle, then find its centroid and display the
position of the centroid inside the triangle.

19. Make a graph with 4 nodes in which every node is connected.

20. For the graph {1 → 2, 2 → 3, 3 → 4, 4 → 1, 3 → 1, 2 → 2}, make a grid giving the
shortest paths between every pair of nodes, with the start node as row and end node
as column.

